

MATEMÀTIQUES

Quadern de pràctiques amb *MAPLE V*

Part II

Montserrat Bruguera i Padró
Albert Ferrer i Biosca
Antoni Guillamon i Grabolosa
Margarida Mitjana i Riera
Francesc Panyella i Brustenga
Xavier Puerta i Coll
Joan J. Rodríguez i Jordana
Carles Serrat i Piè

Professors de la Universitat Politècnica de Catalunya
Departament de Matemàtica Aplicada I
Secció Escola Universitària Politècnica de Barcelona

Barcelona 1999

Aquest quadern de pràctiques amb *MAPLE V* - Part II és continuació del que porta el mateix títol - Part I. Per tant, està destinat a estudiants que ja coneixen el més fonamental d'aquest programa i la seva utilitat en càlculs bàsics d'Àlgebra lineal i Càlcul en una i diverses variables.

La raó per la qual, en el context de l'EUPB, és necessària una ampliació de la primera part, resideix en el fet que les assignatures de Matemàtiques del pla d'estudis d'Enginyeria Tècnica en Topografia contenen matèries, d'interès específic, no contemplades en el primer quadern.

Així, els dos primers capítols, dedicats a les transformacions de semblança, els camps vectorials i la integració en diverses variables, han estat redactats pensant en l'assignatura Mètodes Matemàtics II del segon quadrimestre (1B), i els tres últims, dedicats a les equacions diferencials, la geometria diferencial i les funcions de variable complexa, respectivament, en l'assignatura Ampliació de Càlcul del tercer quadrimestre (2B). En totes aquestes matèries, l'ús d'un assistent de càlcul com *MAPLE V* és especialment recomanable per facilitar la tasca de treball personal necessària per aprofundir en la comprensió dels conceptes matemàtics i la seva utilitat instrumental.

Finalment, val a dir que ambdues parts estan elaborades amb la versió 4 de *MAPLE V* i que, per tant, pot existir alguna diferència de detall amb qualsevol altre versió, tant anterior com posterior.

Barcelona, octubre de 1999

Els Autors.

Índex

1	Transformacions de semblança	1
1.1	Transformacions de semblança bidimensionals	1
1.1.1	Problema directe	1
1.1.2	Problema invers	4
1.2	Transformacions de semblança tridimensionals	9
1.2.1	Problema directe	9
1.2.2	Problema invers	13
1.3	Exercicis	17
2	Camps vectorials i integració en vàries variables	19
2.1	Camps vectorials	19
2.1.1	Camp vectorial de \mathbb{R}^2 en \mathbb{R}^2	19
2.1.2	Camp vectorial de \mathbb{R}^3 en \mathbb{R}^3	21
2.1.3	Camp vectorial de \mathbb{R}^2 en \mathbb{R}^3	22
2.2	Integració en vàries variables	23
2.2.1	Integrals curvilínies de camps vectorials al llarg d'un camí	24
2.2.2	Integrals dobles i triples de camps escalars	27
2.3	Exercicis	31

3	Equacions diferencials	33
3.1	Resolució d'una equació diferencial ordinària.	33
3.2	Condicions inicials. Determinació de les constants.	33
3.3	Representació gràfica de les solucions d'una EDO.	35
3.4	Famílies de corbes ortogonals.	37
3.5	Polinomis de Legendre. Equació de Laplace.	39
3.6	Resolució numèrica d'equacions diferencials.	42
3.7	Problemes proposats	44
4	Geometria diferencial	47
4.1	Definició d'una superfície	47
4.2	Derivades parcials	49
4.3	Punts regulars i singulars d'una superfície	50
4.4	La 1a i 2ona Forma Quadràtica Fonamental	51
4.5	Angle entre corbes	53
4.6	Longitud d'una corba definida sobre una superfície	54
4.7	Àrea d'una superfície	55
4.8	Curvatura normal	56
4.9	Direccions principals i curvatures	56
4.10	Curvatura geodèsica	58
4.11	Projeccions i parametritzacions	59
4.12	Exercicis proposats	61
5	Variable complexa	65
5.1	Operacions elementals amb nombres complexos.	65

5.2	Funcions complexes. Holomorfia i conformitat.	69
5.2.1	Funcions holomorfes	71
5.2.2	Transformacions conformes	72
5.3	Tractament gràfic de corbes paramètriques. Angles entre corbes.	73
5.4	Imatges de dominis per transformacions conformes.	78
5.5	Exercicis proposats.	80

Índex de comandes de MapleV	83
------------------------------------	-----------

Lliçó 1

Transformacions de semblança

Aquest capítol està destinat a mostrar la utilitat que poden tenir els programes de càlcul simbòlic, i en particular MAPLE V, per a la manipulació de transformacions de semblança. El paquet gràfic permet observar els efectes d'una transformació sobre figures de 2 i 3 dimensions, i el paquet d'Àlgebra lineal permet abordar, amb molta facilitat, els càlculs matricials necessaris per a l'estimació dels paràmetres d'una transformació. Tots els procediments s'introduiran a partir d'exemples concrets.

1.1 Transformacions de semblança bidimensionals

1.1.1 Problema directe

En primer lloc es veurà com es pot definir una transformació de semblança bidimensional a partir d'uns valors concrets dels quatre paràmetres que intervenen: angle de rotació α , factor d'escala o raó de la homotècia λ i coordenades del vector de translació $t = (tx, ty)$.

A continuació s'aprofitarà la capacitat gràfica de MAPLE V per a observar els efectes de la transformació sobre algunes figures poligonals.

Carreguem llibreries.

```
> restart:
```

```
> with(linalg):with(plots):with(plottools):
```

```
Warning, new definition for norm
```

Warning, new definition for trace

Donem valors als paràmetres de la transformació.

Rotació de 30° en radians.

```
> alpha:=evalf(30*Pi/180);
      alpha := .5235987758
```

Factor d'escala.

```
> lambda:=2.5;
      lambda := 2.5
```

Vector de translació.

```
> tx:=15:ty:=17: t:=vector(2,[tx,ty]);
      t := [15, 17]
```

Matriu de la homotècia de raó λ composta amb la rotació d'eixos d'angle α .

```
> a:=lambda*cos(alpha): b:=lambda*sin(alpha): A:=matrix(2,2,[a,b,-b,a]);
      A := 
$$\begin{bmatrix} 2.165063509 & 1.250000001 \\ -1.250000001 & 2.165063509 \end{bmatrix}$$

```

construïm la transformació.

```
> v:=vector(2,[x,y]): T:=(v)->evalm(A&*v+t);
      T := v → evalm((A'&*v) + t)
```

Observem l'efecte sobre un segment. Definim els extrems, busquem la seva imatge per la transformació i dibuixem els dos segments.

```
> v1:=vector(2,[3,3]): v2:=vector(2,[9,6]):
plp:=polygonplot([v1,v2],xtickmarks=3,ytickmarks=3):
txp:=textplot([[v1[1],v1[2]],[v2[1],v2[2]]],['A'],['B']]):
vt1:=T(v1): vt2:=T(v2):
plp2:=polygonplot([vt1,vt2],xtickmarks=3,ytickmarks=3):
txp2:=textplot([[vt1[1],vt1[2]],[vt2[1],vt2[2]]],['TA'],['TB']]):
display({plp,txp,plp2,txp2});
      vt1 := [25.24519053, 19.74519052]
      vt2 := [41.98557159, 18.74038104]
```




Figura 1.1: Transformació de semblança

Observem que una rotació d'eixos d'angle α actua com una rotació d'angle $-\alpha$.

D'altra banda, per apreciar millor l'efecte de la transformació, és millor que l'escala sigui la mateixa en els dos eixos de coordenades. Això s'aconsegueix fàcilment clicant sobre la figura i, en l'opció de menú "Projection", escollir "Constrained".

Afegim un tercer punt i observem l'efecte de la transformació sobre un triangle.

```
> v3:=vector(2,[6,9]):
plp:=polygonplot([v1,v2,v3],xtickmarks=3,ytickmarks=3):
txp:=textplot([
[v1[1],v1[2],'A'],
[v2[1],v2[2],'B'],
[v3[1],v3[2],'C']]):
vt3:=T(v3);
plp2:=polygonplot(
[vt1,vt2,vt3],xtickmarks=3,ytickmarks=3):
txp2:=textplot([
[vt1[1],vt1[2],'TA'],
[vt2[1],vt2[2],'TB'],
[vt3[1],vt3[2],'TC']]):
display({plp,txp,plp2,txp2});
```

```
vt3 := [39.24038106, 28.98557157]
```





Figura 1.2: Transformació de semblança

1.1.2 Problema invers

Es tracta d'estimar el valor dels paràmetres de la transformació (angle de rotació α , factor d'escala o raó de la homotècia λ i coordenades del vector de translació $t = (tx, ty)$) a partir de les coordenades d'uns quants punts, anomenats punts de control, i les seves imatges.

Per cada punt de control s'obtenen dues equacions no lineals en els 4 paràmetres α , λ , tx i ty com a incògnites.

$$\lambda \begin{pmatrix} \cos \alpha & \sin \alpha \\ -\sin \alpha & \cos \alpha \end{pmatrix} \begin{pmatrix} x \\ y \end{pmatrix} + \begin{pmatrix} tx \\ ty \end{pmatrix} = \begin{pmatrix} x' \\ y' \end{pmatrix}$$

Aquestes equacions es linealitzen fent el canvi

$$\begin{aligned} a &= \lambda \cos(\alpha) \\ b &= \lambda \sin(\alpha) \end{aligned}$$

Amb dos punts de control s'obté un sistema determinat, però la seva solució és massa dependent dels errors en la determinació de les coordenades d'aquests punts i les seves imatges. Per això, es treballa amb sistemes sobredeterminats corresponents a un mínim de tres punts de control.

Considerem els tres punts v_1 , v_2 i v_3 i uns valors (en general aproximats) dels seus transformats

```
> vt1:=vector(2,[25.2, 19.7]):
vt2:=vector(2,[41.9, 18.7]):
vt3:=vector(2,[39.3, 29.0]):
```

construïm la matriu del sistema sobredeterminat.

```
> A:=matrix(6,4,[
v1[1],v1[2],1,0,v1[2],-v1[1],0,1,
v2[1],v2[2],1,0,v2[2],-v2[1],0,1,
v3[1],v3[2],1,0,v3[2],-v3[1],0,1]);
```

$$A := \begin{bmatrix} 3 & 3 & 1 & 0 \\ 3 & -3 & 0 & 1 \\ 9 & 6 & 1 & 0 \\ 6 & -9 & 0 & 1 \\ 6 & 9 & 1 & 0 \\ 9 & -6 & 0 & 1 \end{bmatrix}$$

Terme independent del sistema sobredeterminat: vector format per les imatges dels punts de control.

```
> U:=( [vt1[1],vt1[2],vt2[1],vt2[2],vt3[1],vt3[2] ] );
      U := [25.2, 19.7, 41.9, 18.7, 39.3, 29.0]
```

A causa d'haver utilitzat valors inexactes dels vectors transformats, el sistema lineal sobredeterminat $Ax = U$ de 6 equacions amb les 4 incògnites a, b, tx, ty és, en principi, incompatible. És a dir, no existeix cap x que faci $Ax - U = 0$. Tanmateix, la funció `leastsqrs` de MAPLE V permet trobar l'anomenada solució en el sentit mínimo-quadràtic, consistent en el vector x que fa $\|Ax - U\|$ mínima.

```
> x:=leastsqrs(A, U);
      x := [2.166666662, 1.258333330, 14.91666672, 17.01666668]
```

Aquest són els valors estimats de les 4 incògnites a, b, tx, ty segons el criteri dels mínims quadrats.

Busquem els valors dels paràmetres λ i α desfent el canvi:

$$\lambda = \sqrt{a^2 + b^2}$$

$$\alpha = \arctan \frac{b}{a}$$

```
> lambda:=sqrt(x[1]^2+x[2]^2); alpha:=arctan(x[2]/x[1]);
      lambda := 2.505563249
      alpha := .5261592004
```

Busquem l'angle de rotació α en graus, minuts i segons sexagesimals.

```
> evalf(alpha*180/Pi);
30.14670153

> (%-30)*60;
8.802092

> (%-8)*60;
48.125520
```

Observem que, per haver utilitzat valors aproximats dels vectors transformats, el vector de translació té una inexactitud de l'ordre de les centèsimes d'unitat, el factor d'escala de l'ordre de les mil·lèsimes i l'angle de rotació d'aproximadament 9 minuts d'arc.

Vegem, ara, què passa si estimem el valor dels paràmetres resolent el sistema determinat que s'obté amb només els dos punts de control A i B.

```
> A1:=matrix(4,4,[
v1[1],v1[2],1,0,v1[2],-v1[1],0,1,
v2[1],v2[2],1,0,v2[2],-v2[1],0,1]);
A1 := 
$$\begin{bmatrix} 3 & 3 & 1 & 0 \\ 3 & -3 & 0 & 1 \\ 9 & 6 & 1 & 0 \\ 6 & -9 & 0 & 1 \end{bmatrix}$$


> U1:=( [vt1[1],vt1[2],vt2[1],vt2[2]] );
U1 := [25.2, 19.7, 41.9, 18.7]

> x1:=evalm(inverse(A1)&*U1);
x1 := [2.160000000, 1.246666666, 14.98000000, 16.96000000]

> lambda:=sqrt(x1[1]^2+x1[2]^2); alpha:=arctan(x1[2]/x1[1]);
lambda := 2.493948230
alpha := .5234564321

> evalf(alpha*180/Pi);
29.99184431
```

```
> (%-29)*60;
                    59.510659
```

```
> (%-59)*60;
                    30.639540
```

I ara amb els dos punts de control B i C.

```
> A1:=matrix(4,4,[
v2[1],v2[2],1,0,v2[2],-v2[1],0,1,
v3[1],v3[2],1,0,v3[2],-v3[1],0,1]);
```

$$A1 := \begin{bmatrix} 9 & 6 & 1 & 0 \\ 6 & -9 & 0 & 1 \\ 6 & 9 & 1 & 0 \\ 9 & -6 & 0 & 1 \end{bmatrix}$$

```
> U1:=( [vt2[1],vt2[2],vt3[1],vt3[2]] );
                    U1 := [41.9, 18.7, 39.3, 29.0]
```

```
> x1:=evalm(inverse(A1)*U1);
                    x1 := [2.150000001, 1.283333335, 14.85000000, 17.35000000]
```

```
> lambda:=sqrt(x1[1]^2+x1[2]^2); alpha:=arctan(x1[2]/x1[1]);
                    λ := 2.503885871
                    α := .5381363996
```

```
> evalf(alpha*180/Pi);
                    30.83294449
```

```
> (%-30)*60;
                    49.976669
```

```
> (%-49)*60;
                    58.600140
```

Observem una diferència considerable segons la parella de punts de control utilitzada. Això és a causa de les diferents aproximacions en els valors de les coordenades dels punts de control transformats. És lògic, doncs, pensar que la millor solució és la donada en el sentit mínim-quadràtic. És l'obtinguda a partir de tota la informació que es disposa i és d'esperar que errors per defecte en les dades compensin altres errors per excés.

Vegem, ara, l'efecte de la transformació, amb els paràmetres estimats per mínims quadrats, sobre els punts de control.

Donem valors als paràmetres de la transformació.

Rotació de $30^{\circ}8'48''$ en radians (precisió de segons).

```
> alpha:=evalf((30+8/60+48/3600)*Pi/180);
      alpha := .5261585920
```

Factor d'escala amb 1 decimal de precisió.

```
> lambda:=2.5;
      lambda := 2.5
```

Vector de translació amb 2 decimals de precisió. Si les unitats són metres, amb precisió de centímetres.

```
> tx:=14.92:ty:=17.02: t:=vector(2,[tx,ty]);
      t := [14.92, 17.02]
```

Matriu de la homotècia de raó λ composta amb la rotació d'eixos d'angle α .

```
> a:=lambda*cos(alpha): b:=lambda*sin(alpha): A:=matrix(2,2,[a,b,-b,a]);
      A := 
$$\begin{bmatrix} 2.161856649 & 1.255538064 \\ -1.255538064 & 2.161856649 \end{bmatrix}$$

```

construïm la transformació.

```
> v:=vector(2,[x,y]): T:=(v)->evalm(A&*v+t);
      T := v  $\rightarrow$  evalm((A'&*v) + t)
```

Vegem l'efecte sobre els tres punts de control, v_1, v_2, v_3 , comparant els valors transformats amb els paràmetres estimats, $T(v_1), T(v_2), T(v_3)$, amb els vectors imatge, vt_1, vt_2, vt_3 , donats com a dada per a estimar els paràmetres.

```
> T(v1);evalm(vt1); T(v2);evalm(vt2); T(v3);evalm(vt3);
      [25.17218414, 19.73895576]
      [25.2, 19.7]
      [41.90993822, 18.69129731]
      [41.9, 18.7]
      [39.19098247, 28.94348146]
      [39.3, 29.0]
```

Observem que, si les coordenades estan en metres, la precisió és de l'ordre dels decímetres.

1.2 Transformacions de semblança tridimensionals

1.2.1 Problema directe

En primer lloc es construirà una transformació de semblança tridimensional a partir d'uns valors concrets dels set paràmetres que intervenen: els tres angles de rotació ω, ϕ, κ , el factor d'escala λ i les coordenades del vector de translació $t = (tx, ty, tz)$.

A continuació s'aprofitarà la capacitat gràfica de MAPLE V per a observar els efectes de la transformació sobre algunes figures poligonals i polièdriques.

Donem valors als paràmetres de la transformació.

Rotació de 30° al voltant de l'eix x , de 15° al voltant de l'eix y i de 45° al voltant de l'eix z . Tots tres angles s'han d'entrar en radians.

```
> omega:=evalf(30*Pi/180);
   phi:=evalf(15*Pi/180);
   kappa:=evalf(45*Pi/180);
       $\omega := .5235987758$ 
       $\phi := .2617993878$ 
```

$$\kappa := .7853981635$$

Factor d'escala.

```
> lambda:=2;
```

$$\lambda := 2$$

Vector de translació.

```
> tx:=5:ty:=7:tz:=10:t:=vector(3,[tx,ty,tz]);
```

$$t := [5, 7, 10]$$

Matriu de la rotació d'eixos d'angles ω , ϕ , κ .

```
> R:=matrix(3,3,[
cos(phi)*cos(kappa),
cos(omega)*sin(kappa)+sin(omega)*sin(phi)*cos(kappa),
sin(omega)*sin(kappa)-cos(omega)*sin(phi)*cos(kappa),
-cos(phi)*sin(kappa),
cos(omega)*cos(kappa)-sin(omega)*sin(phi)*sin(kappa),
sin(omega)*cos(kappa)+cos(omega)*sin(phi)*sin(kappa),
sin(phi),
-sin(omega)*cos(phi),
cos(omega)*cos(phi)]);
```

$$R := \begin{bmatrix} .6830127018 & .7038787867 & .1950597418 \\ -.6830127020 & .5208660846 & .5120470397 \\ .2588190451 & -.4829629133 & .8365163037 \end{bmatrix}$$

construïm la transformació.

```
> v:=vector(3,[x,y,z]):
```

```
T:=(v)->evalm(lambda*(R&*v)+t);
```

$$T := v \rightarrow \text{evalm}(\lambda(R' \& *' v + t))$$

Observem l'efecte sobre un segment. Definim els extrems, busquem els seus transformats, i dibuixem els dos segments.

```
> v1:=vector(3,[6,3,1]): v2:=vector(3,[7,9,2]):
plp:=polygonplot3d([v1,v2],tickmarks=[3,3,3], view=[0..30,0..30,0..30]):
txp:=textplot3d([[v1[1],v1[2],v1[3]],[v2[1],v2[2],v2[3]]],['A'],['B']]):
vt1:=T(v1): vt2:=T(v2):
plp2:=polygonplot3d([vt1,vt2],tickmarks=[3,3,3], view=[0..30,0..30,0..30]):
txp2:=textplot3d([[vt1[1],vt1[2],vt1[3]],[vt2[1],vt2[2],vt2[3]]],['TA'],['TB']]):
display3d({plp,txp,plp2,txp2});
```




Figura 1.3: Transformació de semblança tridimensional

Per apreciar millor l'efecte de la transformació cal clicar sobre la figura i arrossegar el ratolí fins a obtenir el punt de vista adient.

Afegim un tercer punt i observem l'efecte de la transformació sobre un triangle.

```
> v3:=vector(3,[4,5,1]):vt3:=T(v3):
plp:=polygonplot3d([v1,v2,v3],
tickmarks=[3,3,3],view=[0..30,0..30,0..30]):
txp:=textplot3d([
[v1[1],v1[2],v1[3],'A'],
[v2[1],v2[2],v2[3],'B'],
[v3[1],v3[2],v3[3],'C']]):
plp2:=polygonplot3d([vt1,vt2,vt3],
tickmarks=[3,3,3],view=[0..30,0..30,0..30]):
txp2:=textplot3d([
[vt1[1],vt1[2],vt1[3],'TA'],
[vt2[1],vt2[2],vt2[3],'TB'],
[vt3[1],vt3[2],vt3[3],'TC']]):
display3d({plp,txp,plp2,txp2});
```

Afegim un quart punt i observem l'efecte de la transformació sobre un tetraedre.

```
> v4:=vector(3,[5,4,8]):vt4:=T(v4):
plp:=polygonplot3d([v1,v2,v3,v1,v4,v3,v4,v3,v2,v1,v4,v2],
tickmarks=[3,3,3],view=[0..30,0..30,0..30]):
txp:=textplot3d([
[v1[1],v1[2],v1[3],'A'],
[v2[1],v2[2],v2[3],'B'],
[v3[1],v3[2],v3[3],'C'],
[v4[1],v4[2],v4[3],'D']]):
plp2:=polygonplot3d([vt1,vt2,vt3,vt1,vt4,vt3,vt4,vt3,vt2,vt1,vt4,vt2],
tickmarks=[3,3,3],view=[0..30,0..30,0..30]):
```





Figura 1.4: Transformació de semblança tridimensional

```
txp2:=textplot3d([  
[vt1[1],vt1[2],vt1[3], 'TA'],  
[vt2[1],vt2[2],vt2[3], 'TB'],  
[vt3[1],vt3[2],vt3[3], 'TC'],  
[vt4[1],vt4[2],vt4[3], 'TD']]):  
display3d({plp,txp,plp2,txp2});
```




Figura 1.5: Transformació de semblança tridimensional

1.2.2 Problema invers

Es tracta de calcular els valors dels paràmetres d'una transformació de semblança tridimensional conegudes les imatges d'uns quants punts de control.

En el cas bidimensional hem vist que això es fa, en general, resolent un sistema lineal sobredeterminat segons el criteri dels mínims quadrats; però les transformacions de semblança tridimensionals no es poden linealitzar respecte els paràmetres amb un canvi de variable tal i com es fa en el cas bidimensional. Això comporta que el procés d'estimació d'aquests paràmetres a partir de les coordenades d'uns quants punts de control es compliqui bastant i s'hagi de fer per aproximacions successives.

Tanmateix, si la rotació d'eixos correspon a angles α suficientment petits per que valgui l'aproximació $\sin \alpha \simeq \alpha$, $\sin^2 \alpha \simeq 0$ i $\cos \alpha \simeq 1$, i el factor d'escala és molt proper a 1 i es pot escriure de la forma $\lambda = 1 + \varepsilon$, on $\varepsilon^2 \simeq 0$ i $\varepsilon \alpha \simeq 0$, aleshores la transformació es pot escriure de la forma

$$\begin{aligned}\lambda x + \kappa y - \phi z + tx &= x' \\ -\kappa x + \lambda y + \omega z + ty &= y' \\ \phi x - \omega y + \lambda z + tz &= z'\end{aligned}$$

que és lineal respecte els paràmetres $\omega, \phi, \kappa, tx, ty, tz, \lambda$ amb matriu

```
> Alin:=matrix(3,7,[ 0,-z,y,1,0,0,x, z,0,-x,0,1,0,y, -y,x,0,0,0,1,z]);
```

$$A_{lin} := \begin{bmatrix} 0 & -z & y & 1 & 0 & 0 & x \\ z & 0 & -x & 0 & 1 & 0 & y \\ -y & x & 0 & 0 & 0 & 1 & z \end{bmatrix}$$

i amb el vector de paràmetres $par = (\omega, \phi, \kappa, tx, ty, tz, \lambda)$ com a incògnita, i el vector imatge del punt de control ($P' = (x', y', z')$) com a terme independent.

Per a cada punt de control s'obtenen, doncs, 3 equacions lineals en els 7 paràmetres com a incògnites. Amb 2 punts de control el sistema de 6 equacions i 7 incògnites que s'obté és indeterminat i s'haurà de treballar amb sistemes sobredeterminats obtinguts a partir d'un mínim de 3 punts de control.

Vegem un exemple amb 4 punts de control

```
> v1:=vector(3,[6,3,1]);v2:=vector(3,[7,9,2]);
v3:=vector(3,[4,5,1]);v4:=vector(3,[5,4,8]);
```

$$v1 := [6, 3, 1]$$

$$v2 := [7, 9, 2]$$

$$v3 := [4, 5, 1]$$

$$v4 := [5, 4, 8]$$

d'imatges respectives

```
> vt1:=vector(3,[11.14,10.02,11.02]);vt2:=vector(3,[12.21,16.13,12.01]);
vt3:=vector(3,[9.12,12.07,11.00]);vt4:=vector(3,[10.11,11.08,18.15]);
vt1 := [11.14, 10.02, 11.02]
vt2 := [12.21, 16.13, 12.01]
vt3 := [9.12, 12.07, 11.00]
vt4 := [10.11, 11.08, 18.15]
```

construïm les matrius parcials per a cada punt de control i la matriu corresponent als 4 punts

```
> A1:=subs(x=v1[1],y=v1[2],z=v1[3],evalm(Alin)):
A2:=subs(x=v2[1],y=v2[2],z=v2[3],evalm(Alin)):
A3:=subs(x=v3[1],y=v3[2],z=v3[3],evalm(Alin)):
A4:=subs(x=v4[1],y=v4[2],z=v4[3],evalm(Alin)):
A:=stackmatrix(A1,A2,A3,A4);
```

$$A := \begin{bmatrix} 0 & -1 & 3 & 1 & 0 & 0 & 6 \\ 1 & 0 & -6 & 0 & 1 & 0 & 3 \\ -3 & 6 & 0 & 0 & 0 & 1 & 1 \\ 0 & -2 & 9 & 1 & 0 & 0 & 7 \\ 2 & 0 & -7 & 0 & 1 & 0 & 9 \\ -9 & 7 & 0 & 0 & 0 & 1 & 2 \\ 0 & -1 & 5 & 1 & 0 & 0 & 4 \\ 1 & 0 & -4 & 0 & 1 & 0 & 5 \\ -5 & 4 & 0 & 0 & 0 & 1 & 1 \\ 0 & -8 & 4 & 1 & 0 & 0 & 5 \\ 8 & 0 & -5 & 0 & 1 & 0 & 4 \\ -4 & 5 & 0 & 0 & 0 & 1 & 8 \end{bmatrix}$$

Terme independent

```
> vt:=convert(stackmatrix(vt1,vt2,vt3,vt4),vector);
vt := [11.14, 10.02, 11.02, 12.21, 16.13, 12.01, 9.12, 12.07, 11.00, 10.11, 11.08, 18.15]
```

Estimació dels paràmetres, a partir del sistema sobredeterminat de 12 equacions i 7 incògnites, segons el criteri dels mínims quadrats.

```
> par:=leastsqrs(A,vt);

par := [.0051922225, .002971425, .0085343128, 5.000950123, 7.003119366,
        9.996920506, 1.019665274]
```

Angles de rotació en graus minuts i segons sexagesimals

```
> omega:=evalf(par[1]*180/Pi); (%-floor(%))*60; (%-floor(%))*60;
phi:=evalf(par[2]*180/Pi); (%-floor(%))*60; (%-floor(%))*60;
kappa:=evalf(par[3]*180/Pi); (%-floor(%))*60; (%-floor(%))*60;

omega := .2974924355
        17.84954613
        50.972768

phi := .1702501116
        10.21500670
        12.9004020

kappa := .4889801044
        29.33880626
        20.328376
```

Vector de translació

```
> t:=vector(3,[par[4],par[5],par[6]]);
t := [5.000950123, 7.003119366, 9.996920506]
```

Factor d'escala

```
> lambda:=par[7];

lambda := 1.019665274
```

Vegem, ara, l'efecte de la transformació, amb els paràmetres estimats per mínims quadrats, sobre els punts de control.

Donem valors als paràmetres de la transformació.

Entrem els angles de rotació en radians (amb precisió de segons).

```
> omega:=evalf((17/60+51/3600)*Pi/180);
phi:=evalf((10/60+13/3600)*Pi/180);
kappa:=evalf((29/60+20/3600)*Pi/180);

       $\omega := .005192354526$ 

       $\phi := .002971907866$ 

       $\kappa := .008532720790$ 
```

Factor d'escala amb 2 decimals de precisió.

```
> lambda:=1.02;

       $\lambda := 1.02$ 
```

Vector de translació amb 2 decimals de precisió. Si les unitats són metres, amb precisió de centímetres.

```
> tx:=5.00:ty:=7.00:tz:=10.00:t:=vector(3,[tx,ty,tz]);
       $t := [5.00, 7.00, 10.00]$ 
```

Matriu de la rotació d'eixos d'angles ω , ϕ , κ .

```
> R:=matrix(3,3,[
cos(phi)*cos(kappa),
cos(omega)*sin(kappa)+sin(omega)*sin(phi)*cos(kappa),
sin(omega)*sin(kappa)-cos(omega)*sin(phi)*cos(kappa),
-cos(phi)*sin(kappa),
cos(omega)*cos(kappa)-sin(omega)*sin(phi)*sin(kappa),
sin(omega)*cos(kappa)+cos(omega)*sin(phi)*sin(kappa),
sin(phi),
-sin(omega)*cos(phi),
cos(omega)*cos(phi)]);

       $R := \begin{bmatrix} .9999591807 & .008547932774 & -.002927451068 \\ -.008532579569 & .9999499852 & .005217499949 \\ .002971903491 & -.005192308265 & .9999821038 \end{bmatrix}$ 
```

construïm la transformació.

```
> v:=vector(3,[x,y,z]):
T:=(v)->evalm(R&*v+t);
      
$$T := v \rightarrow \text{evalm}((R' \& *' v) + t)$$

```

Vegem l'efecte sobre els quatre punts de control, $v1, v2, v3, v4$, comparant els valors transformats amb els paràmetres estimats, $T(v1), T(v2), T(v3), T(4)$, amb els vectors imatge, $vt1, vt2, vt3, vt4$, donats com a dada per a estimar els paràmetres.

```
> T(v1);evalm(vt1);
T(v2);evalm(vt2);
T(v3);evalm(vt3);
T(v4);evalm(vt4);
      [11.02247143, 9.953871979, 11.00223660]
      [11.14, 10.02, 11.02]
      [12.07079076, 15.95025681, 11.97403676]
      [12.21, 16.13, 12.01]
      [9.039648936, 11.97083711, 10.98590818]
      [9.12, 12.07, 11.00]
      [10.01056803, 10.99887704, 17.99394711]
      [10.11, 11.08, 18.15]
```

Observem que, si les coordenades estan en metres, la precisió és de l'ordre dels centímetres. No es pot obtenir una major precisió quan els angles de rotació estimats són de l'ordre dels minuts d'arc, i s'ha fet una aproximació del tipus $\sin \alpha = \alpha$.

1.3 Exercicis

- Construiu una transformació de semblança bidimensional escollint l'angle de gir, el factor d'escala i el vector de translació.
 - Escolliu tres punts no alineats del pla i busqueu la seva imatge per la transformació.
 - Representeu gràficament els dos triangles.

- Resoleu el problema invers; és a dir, estimeu els paràmetres de la transformació de semblança corresponents als tres punts de control que heu triat.
- 2.
- Construïu una transformació de semblança tridimensional escollint els tres angles de rotació, el factor d'escala i el vector de translació.
 - Escolliu tres punts no alineats del pla i busqueu la seva imatge per la transformació.
 - Representeu gràficament els dos triangles.
 - Varieu els paràmetres i observeu l'efecte de la variació en els triangles transformats.

Lliçó 2

Camps vectorials i integració en vàries variables

En primer lloc es farà un breu recorregut per algunes de les eines principals que disposa MAPLE V per a la manipulació de camps vectorials, incloent la integració al llarg d'un camí. Després es veurà com utilitzar MAPLE V per a calcular integrals dobles i triples de camps escalars. Tots els procediments s'introduiran a partir d'exemples concrets.

S'han de carregar les llibreries `linalg` i `plots`.

2.1 Camps vectorials

Es consideraran els següents conceptes.

- Definir un camp vectorial
- Representar-lo gràficament
- Calcular el seu jacobià
- Preguntar si és conservatiu i, en cas afirmatiu, buscar la funció potencial

2.1.1 Camp vectorial de \mathbb{R}^2 en \mathbb{R}^2

En primer lloc definirem un camp vectorial


```
> F:=vector(2,[2*x*y^2,3+2*x^2*y]);
      F := [2 x y^2, 3 + 2 x^2 y]
```

Per a representar-lo gràficament, utilitzarem la funció `fieldplot` del paquet `plots` que dibuixa, en cada nus (x, y) d'una certa malla que es pot modular amb l'opció `grid`, el corresponent vector $F(x, y)$.

```
> fieldplot(F,x=-2..2, y=-2..2, grid=[10,10]);
```




Figura 2.1: Camp vectorial

Calculem la matriu diferencial o jacobiana del camp vectorial amb la funció `jacobian` del paquet `linalg`.

```
> J:=jacobian(F,[x,y]);
```

$$J := \begin{bmatrix} 2y^2 & 4xy \\ 4xy & 2x^2 \end{bmatrix}$$

Avaluem el jacobiana en un punt. Per exemple, en el punt $P = (1, 2)$.

```
> subs(x=1,y=2,evalm(J));
```

$$\begin{bmatrix} 8 & 8 \\ 8 & 2 \end{bmatrix}$$

La funció `potential` del paquet `linalg` proporciona la funció potencial d'un camp conservatiu, en cas de que ho sigui.

```
> potential(F, [x,y], 'U');
      true
```

En la variable U , que és de sortida, hi ha emmagatzemada la funció potencial:

```
> U;
      x2 y2 + 3 y
```

Vegem que el gradient de U és, efectivament, F .

```
> grad(U, [x,y]);
      [2 x y2, 3 + 2 x2 y]
```

Si el camp no és conservatiu, la funció `potential` ens ho diu. Per exemple,

```
> F2:=vector(2, [x+y, -x-y]): potential(F2, [x,y], 'U2');
      false
```

```
> U2;
      U2
```

2.1.2 Camp vectorial de \mathbb{R}^3 en \mathbb{R}^3

No hi ha cap diferència de procediment respecte el cas bidimensional. Vegem-ho amb un exemple.

```
> F:=vector(3, [3*y2+4*x*z, 6*x*y+2*y*z2, 2*x2+2*y2*z]);
      F := [3 y2 + 4 x z, 6 x y + 2 y z2, 2 x2 + 2 y2 z]
```

```
> fieldplot3d(F, x=-2..2, y=-2..2, z=-2..2, grid=[5,5,5]);
```

```
> J:=jacobian(F, [x,y,z]);
```

$$J := \begin{bmatrix} 4 z & 6 y & 4 x \\ 6 y & 6 x + 2 z^2 & 4 y z \\ 4 x & 4 y z & 2 y^2 \end{bmatrix}$$




Figura 2.2: Camp vectorial

```

> subs(x=1,y=2,z=3,evalm(J));
      [ 12  12   4 ]
      [ 12  24  24 ]
      [  4  24   8 ]

> potential(F, [x,y,z], 'U');
      true

> U;
      3xy2 + 2x2z + y2z2

> grad(U, [x,y,z]);
      [3y2 + 4xz, 6xy + 2yz2, 2x2 + 2y2z]

```

2.1.3 Camp vectorial de \mathbb{R}^2 en \mathbb{R}^3

En aquest cas, la interpretació geomètrica usual és la d'una superfície parametritzada. Per tant, la seva representació gràfica es farà amb `plot3d` en comptes de `fieldplot`, i no té sentit buscar-li funció potencial. Vegem, com a exemple, la parametrització de la catenoide, superfície que s'obté per rotació de la catenària.

```

> F:=vector(3,[cos(u)*cosh(v), sin(u)*cosh(v), v]);
      F := [cos(u) cosh(v), sin(u) cosh(v), v]

```

Representació gràfica

```
> plot3d(F,u=-0..2*Pi,v=-2..2);
```




Figura 2.3: Catenoide

Càlcul del jacobiana

```
> jacobian(F,[u,v]);
```

$$\begin{bmatrix} -\sin(u) \cosh(v) & \cos(u) \sinh(v) \\ \cos(u) \cosh(v) & \sin(u) \sinh(v) \\ 0 & 1 \end{bmatrix}$$

Per aprofundir en l'estudi de superfícies vegeu el capítol de Geometria Diferencial.

2.2 Integració en vàries variables

Es tracta de fer una introducció a algunes de les eines bàsiques que disposa MAPLE V per al càlcul d'integrals de funcions de vàries variables. Es tractaran els següents casos.

1. Integrals curvilínies de camps vectorials al llarg d'un camí
 - Definir un camp vectorial i representar-lo gràficament
 - Definir un camí d'integració i representar-lo gràficament

- Calcular la integral curvilínia corresponent

2. Integrals dobles i triples de camps escalars

- Definir un camp escalar i, si es tracta d'una funció de dues variables, representar-la gràficament
- Definir i representar gràficament, un recinte d'integració.
- Calcular una integral doble o triple en el recinte donat, en forma d'integrals simples reiterades.

2.2.1 Integrals curvilínies de camps vectorials al llarg d'un camí

Sigui, per exemple, el camp vectorial

```
> F:= vector(2, [-sin(x)*sin(y), cos(x)*cos(y)]);
      F := [-sin(x) sin(y), cos(x) cos(y)]
> fieldplot(F, x=-2*Pi..2*Pi, y=-2*Pi..2*Pi, grid=[15,15]);
```




Figura 2.4: Camp vectorial

Integrem al llarg de la cúbica $y = x^3$, parametritzada per $r(t) = (t, t^3)$, des del punt $A = (-2, -8)$ fins al punt $B = (2, 8)$

```
> r:=vector(2, [t, t^3]); plot([r[1], r[2], t=-2..2]);
      r := [t, t^3]
```

Handwritten notes:
 $r = (t, t^3)$
 $y = t^3$




Figura 2.5: Cúbica

Dibuixem la corba dins el camp vectorial

```
> plfi:=fieldplot(F, x=-2..2, y=-8..8, grid=[20,20]):
plc:=plot([r[1],r[2], t=-2..2]):
display(plfi,plc);
```




Figura 2.6: Cúbica dins el camp vectorial

Substituïm la parametrització de la corba en el camp vectorial. Observi's que, perquè actui la funció subs, hem d'aplicar evalm a F i, per tant, considerar F com a matriu. Si volem que continui sent un vector per poder fer producte escalar, hem d'aplicar la funció

convert.

```
> Ft:=convert(subs(x=t,y=t^3,evalm(F)),vector);
      Ft := [-sin(t) sin(t^3), cos(t) cos(t^3)]
```

La funció integrand *fint* és el producte escalar de F i dr , vector de derivades del camp vectorial r que defineix la corba. S'ha de tenir en compte que, per a derivar un camp vectorial com r , no es pot utilitzar directament la funció `diff` sinó que s'ha de fer a través de la funció `map`.

```
> dr:=map(diff,r,t); fint:=dotprod(Ft,dr);
      dr := [1, 3 t^2]
      fint := -sin(t) sin(t^3) + 3 cos(t) cos(t^3) t^2
```

Calculem la integral de la funció *fint* des de $t = -2$, corresponent al punt A , fins a $t = 2$, corresponent al punt B . Recordeu que la funció `Int` (amb majúscula) serveix, únicament, per a expressar la integral amb la notació habitual. Si es vol calcular efectivament la integral, s'ha de fer servir `int` (amb minúscula).

```
> Int(fint,t=-2..2); evalf(int(fint,t=-2..2));
      ∫-22 -sin(t) sin(t^3) + 3 cos(t) cos(t^3) t^2 dt
      - .8234366091
```

Si F és un camp conservatiu, calcularem la integral avaluant la funció potencial en els punts inicial i final.

```
> potential(F, [x,y], 'U');
      true
> U;
      -1/2 sin(x - y) + 1/2 sin(x + y)
> expand(%);
      cos(x) sin(y)
> evalf(subs(x=2,y=8,U)-subs(x=-2,y=-8,U));
      - .8234366090
```

2.2.2 Integrals dobles i triples de camps escalars

Integrals dobles

Càlcul del volum per sota d'una volta de Viviani.

Considerem l'hemisferi nord de la superfície esfèrica de radi $2R$

```
> esf:=sqrt(4*R^2-x^2-y^2);
```

$$esf := \sqrt{4R^2 - x^2 - y^2}$$

Només per a fer la representació gràfica, considerem el cas particular $R = 1$.

```
> z1:=subs(R=1,esf); plot3d(z1,x=-2..2,y=0..2,axes=normal);
```

$$z1 := \sqrt{4 - x^2 - y^2}$$




Figura 2.7: Esfera

Recinte d'integració: circumferència de radi R (prendrem $R=1$ a efectes de representació gràfica) en el pla $z = 0$ centrada al punt $(0, 1, 0)$.

```
> implicitplot(x^2+(y-1)^2=1,x=-1..1,y=0..2);
```

Per poder dibuixar la superfície i el recinte d'integració junts, hem d'entrar la circumferència com una corba parametritzada en 3D, perquè la funció de MAPLE V display no pot representar alhora gràfics en 2D i gràfics en 3D. L'opció `style = wireframe` és per que no ocultí les línies i la superfície esfèrica no amagui el recinte d'integració.




Figura 2.8: Recinte d'integració

```
> pls:=plot3d(z1,x=-2..2,y=0..2,axes=normal):
plr:=spacecurve([cos(t),1+sin(t),0],t=0..2*Pi):
display(pls,plr,style=wireframe);
```




Figura 2.9: Volta de Viviani

Límits d'integració en coordenades polars r, α .

$$\begin{aligned} 0 &\leq \alpha \leq \pi \\ 0 &\leq r \leq 2R \sin \alpha \end{aligned}$$

Integrant en coordenades polars: substituïm el canvi i multipliquem pel jacobià.

```
> zp:=subs(x=r*cos(alpha),y=r*sin(alpha),esf)*r ;
```

$$zp := \sqrt{4R^2 - r^2 \cos(\alpha)^2 - r^2 \sin(\alpha)^2} r$$

```
> zp:=simplify(%);
```

$$zp := \sqrt{4R^2 - r^2} r$$

Per a fer el càlcul efectiu d'una integral doble, s'ha de fer com integrals simples reiterades. Observi's, de nou, la diferència entre les funcions de MAPLE `V int` i `Int`.

```
> Int(Int(zp,r=0..2*R*sin(alpha)),alpha=0..Pi);
```

$$\int_0^\pi \int_0^{2R \sin(\alpha)} \sqrt{4R^2 - r^2} r \, dr \, d\alpha$$

```
> int(int(zp,r=0..2*R*sin(alpha)),alpha=0..Pi);
```

$$\frac{8}{3} (R^2)^{3/2} \pi - \frac{32}{9} (R^2)^{3/2}$$

```
> factor(simplify(%,assume=positive));
```

$$\frac{8}{9} R^3 (3\pi - 4)$$

La simplificació no hagués estat completa si no s'hagués suposat que el resultat és positiu.

Integrals triples

Calculem, ara, mitjançant una integral triple en coordenades esfèriques, la massa de la porció d'esfera $x^2 + y^2 + z^2 = 18$, amb densitat variable donada pel camp escalar $f(x, y, z) = x + y + z$, situada dins el con $x^2 + y^2 = z^2$, i a l'hemisferi nord.

Definim les dues superfícies i les representem gràficament

```
> z1:=sqrt(18-x^2-y^2); z2:=sqrt(x^2+y^2);
plot3d({z1,z2},x=-4..4,y=-3..3);
```

$$z1 := \sqrt{18 - x^2 - y^2}$$

$$z2 := \sqrt{x^2 + y^2}$$




Figura 2.10: Intersecció d'esfera i con

Intersecció entre les dues superfícies

```
> solve({z=z1,z=z2},x,y,z);
      {x = RootOf(-Z^2 - 9 + y^2), y = y, z = 3}
```

Veiem que la intersecció està en el pla $z = 3$. Si fem $z = 3$ en les equacions de les dues superfícies, comprovarem que es tracta de la circumferència de radi 3, $x^2 + y^2 = 9$, situada en aquest pla. D'aquí es dedueix, també, que la inclinació de la generatriu del con és $\frac{\pi}{4}$.

Límits d'integració en coordenades esfèriques r, θ, ϕ .

$$\begin{aligned} 0 &\leq \theta \leq 2\pi \\ \frac{\pi}{4} &\leq \phi \leq \frac{\pi}{2} \\ 0 &\leq r \leq 2\sqrt{3} \end{aligned}$$

Funció densitat

```
> rho:=x+y+z;
      ρ := x + y + z
```

Funció integrand. Substituïm coordenades esfèriques i multipliquem pel jacobí del canvi

```
> fint:=subs(
x=r*cos(phi)*cos(theta),
y=r*cos(phi)*sin(theta),
z=r*sin(phi),rho)*r^2*cos(phi);
      fint := (r cos(ϕ) cos(θ) + r cos(ϕ) sin(θ) + r sin(ϕ)) r^2 cos(ϕ)
```

Integral triple en coordenades esfèriques com a integrals simples reiterades. Observi's, de nou, la diferència entre les funcions de MAPLE `int` i `Int`.

```
> Int(Int(Int(fint, r=0..sqrt(18)), phi=Pi/4..Pi/2), theta=0..2*Pi);
int(int(int(fint, r=0..sqrt(18)), phi=Pi/4..Pi/2), theta=0..2*Pi);
```

$$\int_0^{2\pi} \int_{1/4\pi}^{1/2\pi} \int_0^{3\sqrt{2}} (r \cos(\phi) \cos(\theta) + r \cos(\phi) \sin(\theta) + r \sin(\phi)) r^2 \cos(\phi) dr d\phi d\theta$$

$$\frac{81}{2} \pi$$

Observi's que la funció

$$\rho = r \cos \phi \cos \theta + r \cos \phi \sin \theta + r \sin \phi,$$

com a funció densitat que és, té dimensions de massa partit per volum i que el diferencial de volum

$$r^2 \cos \phi dr d\phi d\theta$$

té dimensions de volum. Per tant, l'integrand i la integral tenen dimensions de massa.

2.3 Exercicis

1. Donat el camp vectorial $F(x, y) = (xy, x^2 - y^2)$

- Representeu-lo gràficament
- Comproveu si té funció potencial i, en el seu cas, calculeu-la
- Calculeu la integral curvilínia entre els punts $A = (-1, 1)$ i $B = (1, 1)$ al llarg dels segents camins
 - Circumferència centrada a l'origen i de radi $\sqrt{2}$
 - Paràbola $y^2 = x$
 - Corba $y^2 = x^3$
 - el.lipse $x^2 + 2y^2 = 3$

2. Repetiu l'exercici anterior amb el camp vectorial $F(x, y) = (2xy, x^2 - y^2)$

3. Siguin el paraboloides $z = x^2 + y^2$ i el pla $x + y = 1$
- Representeu les dues superfícies gràficament
 - Calculeu el volum limitat per aquestes dues superfícies i els tres plans coordenats
4. Siguin el paraboloides $3z = x^2 + y^2$ i l'esfera $x^2 + y^2 + z^2 = 4$
- Representeu les dues superfícies gràficament
 - Calculeu el volum, interior al paraboloides, limitat per aquestes dues superfícies

Lliçó 3

Equacions diferencials

3.1 Resolució d'una equació diferencial ordinària.

La resolució d'una equació diferencial amb *MAPLE V* no necessita carregar cap llibreria especial. La funcions estàndard `diff` i `dsolve` ens permeten definir l'equació diferencial i calcular la seva solució. Aquesta solució depèn de certes constants que es poden determinar a partir de les condicions inicials. Definim una equació diferencial amb la funció `diff`:

```
> diff(x(t),t)=0.2*x(t);
```

$$\frac{\partial}{\partial t} x(t) = .2x(t)$$

i la resollem utilitzant la funció `dsolve`

```
> dsolve(diff(x(t),t)=.2*x(t),x(t));
```

$$x(t) = e^{(.2000000000 t)} _C1$$

Observeu que la solució és una família de corbes que depèn de la constant $C1$.

3.2 Condicions inicials. Determinació de les constants.

Considerem la família de corbes, que depèn de dos paràmetres $c1$ i $c2$, definida pel procediment `corba`:

```
> corba:=(c1,c2)->(c1+c2*t)*exp(3*t);
      corba := (c1, c2) → (c1 + c2 t) e(3t)
```

Aquestes corbes són solució de l'equació diferencial de segon ordre

$$y'' - 6y' + 9y = 0.$$

En *MAPLE V* aquesta equació diferencial la utilitzarem creant un procediment `ed`:

```
> ed:=y->diff(y,t$2)-6*diff(y,t)+9*y;
      ed := y → (∂2/∂t2 y) - 6 (∂/∂t y) + 9 y
```

Verifiquem que resol l'equació diferencial utilitzant la funció `simplify`

```
> simplify(ed(corba(c1,c2)));
      0
```

En aquest punt calcularem els valors de les constants $c1$ i $c2$ per tal de determinar la corba de la família que satisfà les condicions inicials $y(0.1) = 12$ i $y'(0.1) = 1$. Per resoldre un sistema d'equacions $eq1 = 0$, $eq2 = 0$ s'usa la comanda `solve(eq1,eq2)`.

Observeu la manera de definir les equacions del sistema que volem resoldre fent servir l'assignació `eq1`:

```
> eq1:=subs(t=0.1,corba(c1,c2))-12; eq2:=subs(t=0.1,
diff(corba(c1,c2),t))-1;
      eq1 := (c1 + .1 c2) e3 - 12
      eq2 := c2 e3 + 3(c1 + .1 c2) e3 - 1.
```

Un cop definides les equacions determinem les constants:

```
> solve({eq1,eq2});
      {c1 = 11.48268242, c2 = -25.92863772}
```

3.3 Representació gràfica de les solucions d'una EDO.

Les funcions i procediments de la llibreria `DEtools` d'equacions diferencials de *MAPLE V* es poden utilitzar fent servir la comanda `with(DEtools)`. Aleshores, hi ha tot un seguit de funcions disponibles que podeu consultar amb el `Help` de *MAPLE V*. Inicialitzem la sessió fent servir les comandes que carreguen les llibreries `DEtools` i `plots` d'equacions diferencials i de representació gràfica:

```
> with(DEtools):with(plots):

Warning, new definition for adjoint
```

Repassarem algunes utilitats d'aquestes llibreries en un exemple concret. L'equació diferencial:

```
> diff(x(t),t)=0.2*x(t);
```

$$\frac{\partial}{\partial t} x(t) = .2 x(t)$$

es pot pensar com un model matemàtic de **creixement d'una població**, sense depredadors naturals, al llarg del temps. Aleshores, la variació de la població serà proporcional al nombre d'individus que hi ha en un instant donat ($x'(t) = ax(t)$ amb $a > 0$). La solució d'aquesta equació és una funció exponencial. La representació gràfica de la solució s'obté amb la funció `DEplot`. Comentarem breument aquesta funció que ens permetrà, donada una llista d'equacions diferencials ordinaries, representar gràficament les corbes solució utilitzant mètodes numèrics.


L'expressió general de la funció és:

```
DEplot(deqns, vars, trange, inits, xrange, yrange, eqns) on,
```

```
deqns  -llista d'equacions diferencials de qualsevol ordre
vars   -llista de variables dependents
trange -rang de les variables independents
inits  -condicions inicials
xrange -rang de les primeres variables dependents
yrange -rang de les segones variables dependents
eqns   -altres opcions de la forma keyboard=value
```

En el nostre cas representarem gràficament les corbes de la família de solucions de $x'(t) = ax(t)$ on $a > 0$ amb les condicions inicials $x(0) = 2$ i $x(0) = 0.5$.


```
> DEplot({diff(x(t),t)=.2*x(t)},{x(t)},t=0..10,[[x(0)=2],
[x(0)=0.5]], stepsize=.1,linecolor=[blue,green],arrows=none,
method=rkf45);
```



El model matemàtic de creixement d'una població que no té depredadors naturals no és molt real. Per això cal pensar en els models matemàtics anomenats **sistemes depredador-presa**. A la Natura existeix una competició constant per la supervivència entre les diferents espècies d'animals que viuen en un mateix entorn. Uns animals sobreviuen alimentant-se d'altres, els quals desenvolupen mètodes d'evasió per evitar ser depredats. Les equacions del model matemàtic depredador-presa proposat per V.Volterra són les següents:

$$\begin{aligned}x'(t) &= ax(t) - bx(t)y(t) - ex(t) = x(t)(a - by(t) - e) \\y'(t) &= -cy(t) + dx(t)y(t) - ey(t) = y(t)(dx(t) - c - e).\end{aligned}$$

Podem suposar que tots els coeficients són positius. Observem que $x(t)$ és el nombre de preses i $y(t)$ el de depredadors. L'augment de preses és proporcional a la població mentre que la disminució s'estima a partir del nombre d'encontres que hi ha amb els depredadors, que és proporcional a $x(t)y(t)$, i menys un cert factor de decreixement degut a altres causes. En el cas dels depredadors, la variació respecte el nombre d'individus, és decreixent ja que si no hi hagués preses aquests desapareixerien, però la població augmenta proporcionalment al nombre d'encontres amb les preses.

Les solucions explícites $x(t)$ i $y(t)$ del model de V.Volterra no es poden donar en termes de funcions elementals. Ara bé, podem calcular implícitament la família de solucions $f(x, y, k) = 0$ que depenen d'un paràmetre k i representar-les gràficament. Per a fer la representació gràfica de les solucions del model, definim les assignacions **equacio1** i **equacio2**:

```
> equacio1:=x(t)*(a-b*y(t)-e); equacio2:=y(t)*(c*x(t)-d-e);
```

$$\text{equacio1} := x(t) (a - by(t) - e)$$


$$\text{equacio2} := y(t) (cx(t) - d - e)$$

Amb aquestes assignacions definim el procediment **Lotka Volterra** que, utilitzant com a paràmetres a , b , c , d i e , ens permetrà representar gràficament les solucions.

```
> LotkaVolterra:=(a,b,c,d,e,pint)->DEplot({dif f(x(t),t)=
x(t)*(a-b*y(t)-e), diff(y(t),t)=y(t)*(d*x(t)-c-e)},
[x(t),y(t)],t=0..20,[[x(0)=2,y(0)=2],[x(0)=(c+e)/d,y(0)=(a-e)/b]],
stepsize=.1,linecolor=[pint,pint], arrows=none,method=rkf45):
```

Utilitzant la funció `display` fem la representació gràfica conjunta de les solucions del model de Volterra per a 3 nivells diferents, $e = 0, 0.5, 1.2$.

```
> display({LotkaVolterra(1,1,1,1,0,blue),LotkaVolterra(1,1,
1,1,0.5,green), LotkaVolterra(1,1,1,1,1.2,red)},scaling=constrained,
title='e=0,0.2 i 0.5');
```



3.4 Famílies de corbes ortogonals.

Donada una família de corbes que depèn d'un paràmetre ens podem plantejar el problema de trobar la família de corbes orthogonal a la donada. És a dir, la família de corbes que en cada punt les talla formant un angle recte. Aquest és un problema clàssic d'equacions diferencials. Per resoldre'l amb *MAPLE V*, carreguem la llibreria `plots`

```
> with(plots):
```

El procediment `f` efectua el dibuix de la família de funcions $k(1 - x^2)$ de paràmetre k utilitzant la funció `plot`

```
> f:=k->plot(k*(1-x^2),x=-2..2,discont=true,numpoints=500):
```

Si derivem l'anterior equació podem trobar el paràmetre k en funció de la derivada y' . Substituint a l'equació el valor del paràmetre s'obté l'equació diferencial de la família de corbes que és $y'(1 - x^2) + 2xy = 0$. Si substituïm y' per $-1/y'$ s'obté l'equació diferencial de la família de corbes ortogonal, que és $2xyy' + x^2 - 1 = 0$. La família de corbes solució d'aquesta última equació diferencial ve donada per

$$y^2 - \log|x| + x^2/2 = c \quad (3.1)$$

com es pot veure fent servir `dsolve`:

```
> dsolve(2*x*y(x)*diff(y(x),x)+x^2-1= 0,y(x));
```

$$y(x) = \frac{1}{2} \sqrt{-2x^2 + 4C1 + 4\ln(x)}, y(x) = -\frac{1}{2} \sqrt{-2x^2 + 4C1 + 4\ln(x)}$$


Observeu que la solució obtinguda per `dsolve` és la mateixa que (3.1) multiplicada per una constant.

El procediment `g` efectua el dibuix de la família de funcions ortogonals de paràmetre c . Observeu que aquest procediment utilitza la funció `implicitplot`.

```
> g:=c->implicitplot(y^2-log(abs(x))+x^2/2 = c,x=-5..5,y=-5..5,
numpoints=2000):
```

Dibuixem les corbes de la família original i de l'ortogonal per a uns quants valors dels paràmetres (que no tenen res d'especial).

```
> display(g(.6),g(1),g(2),f(1),f(2),scaling=constrained);
```



3.5 Polinomis de Legendre. Equació de Laplace.

Els polinomis de Legendre es poden obtenir a partir de la fórmula

$$P_n(t) = \frac{1}{2^n n!} \frac{d^n}{dt^n} (t^2 - 1)^n,$$


que s'anomena fórmula de *Rodrigues*. Els polinomis de Legendre tenen, entre d'altres propietats, la d'aproximar funcions a l'interval $[-1, 1]$. Aquesta aproximació és en cert sentit, que no explicitarem, la millor aproximació per polinomis que podem trobar d'una funció a l'interval $[-1, 1]$. La funció **Elgendre** ens permet avaluar el polinomi de Legendre de grau n per la variable x utilitzant la fórmula de Rodrigues.

```
> Elgendre := (x, n) -> subs(t=x, diff((t^2-1)^n, t$n) / 2^n/n!);
```

$$\text{Elgendre} := (x, n) \rightarrow \text{subs}(t = x, \frac{\text{diff}((t^2 - 1)^n, t \$ n)}{2^n n!})$$

Com exemple utilitzant el procediment `plot` podem representar gràficament el polinomi de Legendre de grau 100 a l'interval $[-1, 1]$. Observeu a la gràfica que passa als extrems de l'interval.

```
> plot(Elgendre(x, 100), x=-1..1);
```



Farem servir ale polinomis de Legendre per emprar altres procediments de *MAPLE V*.

Definim la funció **IntLegendre**. Per exemple, si volem calcular la integral del polinomis de Legendre a l'interval $[-1, 1]$:

```
> IntLegendre:=n->int(Elgendre(x,n),x=-1..1);
```

$$IntLegendre := n \rightarrow \int_{-1}^1 Elgendre(x, n) dx$$

D'altra banda, el procediment **collect** permet determinar el resultat d'un càlcul en funció de l'expressió que indiquem, en aquest cas és la variable w . Això ens pot ser útil per definir la funció **LegAmbPes** que permet avaluar polinomis de Legendre multiplicats per una funció (polinomis de Legendre amb pes). En el nostre cas considerem la funció $1 - w^2$.

```
> LegAmbPes:=n->collect(simplify((1-w^2)*Elgendre(w,n)),w);
```

$$LegAmbPes := n \rightarrow \text{collect}(\text{simplify}((1 - w^2) Elgendre(w, n)), w)$$

```
> LegAmbPes(6);
```

$$-\frac{231}{16} w^8 + \frac{273}{8} w^6 - \frac{105}{4} w^4 + \frac{55}{8} w^2 - \frac{5}{16}$$

La funció **IntLegPes** permet calcular, a l'interval $[-1, 1]$, les integrals del polinomis de Legendre amb pes

```
> IntLegPes:=n->int(LegAmbPes(n),w=-1..1);
```

$$IntLegPes := n \rightarrow \int_{-1}^1 LegAmbPes(n) dw$$

Solucions de l'equació de Laplace.

Si tenim l'equació de Laplace

$$\frac{\partial^2 f(x, y)}{\partial x^2} + \frac{\partial^2 f(x, y)}{\partial y^2} = 0$$

i la resollem fent servir el procediment `pdesolve`, s'obté com a solució una funció que és una suma de funcions arbitràries dels nombres complexos $-x + yi$ i $y - xi$

```
> pdesolve(diff(f(x,y),x$2)+diff(f(x,y),y$2)=0, f(x,y));
f(x, y) = _F1(I y - x) + _F2(y - I x)
```

Observeu que, en aquest cas, *MAPLE V* ens dóna una informació poc clara. A partir d'aquest últim resultat i del fet que moltes vegades les condicions de contorn inicials es donen sobre un cercle, sembla natural utilitzar un canvi a coordenades polars, $x(r, \theta) = r \cos(\theta)$ i $y(r, \theta) = r \sin(\theta)$, per representar l'equació de Laplace. L'expressió en polars de l'equació de Laplace és


$$\frac{\partial^2 f}{\partial r^2} + \frac{1}{r} \frac{\partial f}{\partial r} + \frac{1}{r^2} \frac{\partial^2 f}{\partial \theta^2} = 0.$$

Aleshores, se sap que la solució general de l'equació de Laplace es pot expressar en forma de sèrie com

$$f(r, \theta) = \sum_{n=0}^{\infty} a_n r^n P_n(\cos(\theta)),$$

on P_n són els polinomis de Legendre i els coeficients a_n s'han de determinar a partir de les condicions de contorn inicials. El que ens interessa destacar, de l'anterior expressió, és que les funcions $r^n P_n(\cos(t))$ són solucions particulars de l'equació de Laplace. Vegem la representació gràfica d'alguna d'aquestes solucions

```
> plot3d(r^5*Elgendre(cos(t),5),r=0..1,t=0..2*Pi,axes=normal);
```



3.6 Resolució numèrica d'equacions diferencials.

Carreguem les llibreries DEtools, plots i linalg.

```
> with(DEtools):with(plots):with(linalg):
```

Considerem l'equació diferencial que ja hem resolt a la secció 3.1

```
> diff(x(t),t)=.2*x(t);
```

$$\frac{\partial}{\partial t} x(t) = .2x(t)$$

La funció $x(t) = 2e^{0.2t}$ és una solució exacta per a la condició inicial $x(0) = 2$. Definim l'assignació de dibuix `Exacta` de la solució exacta:

```
> Exacta:=plot(exp(0.2*t)*2,t=0..20,color=[black],thickness=3):
```

i els procediments de dibuix Oiler de la solució aproximada pel Mètode d'Euler:

```
> Oiler:=h-> DEplot(diff(x(t),t)=.2*x(t),x(t),t=0..20,[[x(0)=2]],
  stepsize=h ,linecolor=[red],arrows=small,method=classical[foreuler]):
```


i Heun de la solució aproximada pel Mètode d'Euler modificat:

```
> Heun:=h-> DEplot(diff(x(t),t)=.2*x(t),x(t),t=0..20,[[x(0)=2]],
  stepsize=h ,linecolor=[green],arrows=none,method=classical[heunform]):
```

Amb el procediment `display` mostrem totes les gràfiques de cop per diferents valors del pas d'integració h .


Pas d'integració: $h=1$.

```
> display(Exacta,Oiler(1),Heun(1));
```




Pas d'integració: $h=0.5$.

```
> display(Exacta,Oiler(0.5),Heun(0.5));
```



Pas d'integració: $h=0.1$.

```
> display(Exacta,Oiler(0.1),Heun(0.1));
```

Fixeu-vos que com més petit és el pas d'integració, més alta és la precisió de les solucions calculades numèricament.

3.7 Problemes proposats

1. Busqueu les solucions (utilitzant directament *MAPLE V*) fent totes les passes tal com ho feu a classe. Compareu els resultats obtinguts.

$$(a) \quad y' + y = \frac{1}{1 + e^{2x}};$$

$$(b) \quad x y' = y + 2x e^{-y/x};$$

$$(c) \quad \left(x + \frac{2}{y}\right) dy + y dx = 0;$$

$$(d) \quad x y' = 2x + 3y;$$

$$(e) \quad (x y^2 + n x^2 y) dx + (x^3 + x^2 y) dy = 0 \quad (\text{Quant ha de valer } n \text{ perquè sigui exacta?});$$

$$(f) \quad x y' - 3y = x^4;$$

$$(g) \quad 0 = \frac{y}{1 - x^2 - y^2} dx + \frac{x}{1 - x^2 - y^2} dy.$$

2. Donada la família de corbes $k \cdot \sin x$, calculeu l'equació diferencial de la família de corbes ortogonals. Resoleu aquesta equació i feu la representació gràfica conjunta de les dues famílies de corbes.
3. Trobeu la solució particular de la família de corbes solució de les equacions diferencials dels apartats (a), (d) i (f) amb la condició inicial $y(0) = 1.5$.

4. Trobeu, utilitzant el Mètode d'Euler, la solució aproximada de la família de corbes solució de les equacions diferencials dels apartats (a), (d) i (f) amb la condició inicial $y(0) = 1.5$ i compareu gràficament els resultats amb els obtinguts al problema anterior.
5. Feu la representació gràfica conjunta de les corbes solució de l'equació diferencial de l'apartat (d) i la família de corbes ortogonals.
6. **Equació de Legendre. Polinomis de Legendre**
L'equació diferencial de 2on ordre

$$(1 - x^2)y'' - 2xy' + n(n + 1)y = 0 \quad (3.2)$$

s'anomena *equació de Legendre*.

Per al cas $n = 1$ és conegut que les solucions són de la forma

$$y = c_1x + c_2 \left(\frac{x}{2} \log \left(\frac{1+x}{1-x} \right) - 1 \right).$$

Per a $n > 1$, $n \in \mathbb{N}$, les solucions vénen donades pels anomenats *polinomis de Legendre*, que es poden definir de la següent manera:

$$P_n(x) = \frac{1}{2^n n!} \frac{d^n}{dx^n} ((x^2 - 1)^n).$$

Per als nostres propòsits, només ens fixarem en ells en l'interval $[-1, 1]$.

- (a) Dibuixeu els polinomis de Legendre per a $n = 0, \dots, 5$.
- (b) Comproveu, per a $n = 0, \dots, 5$, que els polinomis de Legendre són solucions de l'equació de Legendre (3.2).
- (c) Utilitzeu la comanda `animate3d` per representar els polinomis de Legendre per a $n = 0, 5, 10, \dots, 50$.
- (d) Comproveu que $\int_{-1}^1 P_{2m}(x) dx = 0$ per $m = 0, 1, \dots, 50$.
7. Sigui $V = V(x, y, z)$. Doneu l'expressió de l'equació de Laplace $\Delta V = 0$ en coordenades esfèriques.
8. Doneu una aproximació d'ordre 10 de la solució de l'equació de Laplace $\Delta V = 0$, amb la condició que $V = 1 - \cos(\theta)$ sobre els punts de l'esfera unitat, essent θ la latitud del punt.

Lliçó 4

Geometria diferencial

En aquesta secció es donen un conjunt de definicions genèriques que permeten calcular la 1a i 2a formes quadràtiques fonamentals, el vector normal unitari en un punt, i altres elements associats a superfícies i corbes sobre superfícies. Aquestes definicions s'apliquen com a exemple, per a fer els càlculs corresponents sobre una superfície concreta.

En cada secció s'explica la forma de calcular un element de la superfície i en cada apartat hi ha un exemple d'aplicació.

Primer de tot cal inicialitzar el sistema i carregar les llibreries que facin falta.

```
> restart:with(plots):with(linalg):
```

```
Warning, new definition for norm
```

```
Warning, new definition for trace
```

Les superfícies *esfera*, *parabhiper*, i Φ que es defineixen en la secció següent es faran servir durant tot el capítol.

4.1 Definició d'una superfície

En la secció 4.2.2 de la part I del *Quadern de Pràctiques amb MAPLE V*, s'han representat gràficament algunes superfícies que s'han definit prèviament. Amb l'objectiu de calcular elements característics de les superfícies parametritzades, les definirem com funcions vectorials de dues variables. Per exemple, l'esfera

```
> esfera:=(u,v)->[R*cos(v)*cos(u),R*cos(v)*sin(u),R*sin(v)]:
```

la banda de Möbius

```
> mobius:=(u,v)->[(2-v*sin(u/2))*sin(u),
(2-v*sin(u/2))*cos(u),v*cos(u/2)]:
```

un paraboloid hiperbòlic

```
> parabhiper:=(u,v)->[u+v,u-v,u*v];
parabhiper := (u, v) → [u + v, u - v, u v]
```


o una superfície $\Phi(u, v)$

```
> Phi:=(u,v)->[exp(u)*cos(v),exp(u)*sin(v),(u+1)^2];
Φ := (u, v) → [eu cos(v), eu sin(v), (u + 1)2]
```

Representem gràficament aquestes dues darreres funcions ja que per les dues primeres n'havíem obtingut les gràfiques en la secció 4.2.2 de la part I del *Quadern de Pràctiques amb MAPLE V*.


La superfície Φ

```
> plot3d([exp(u)*cos(v),exp(u)*sin(v),(u+1)^2],u=-2..5,
v=-Pi..Pi,grid=[40,40],axes=framed,title='phi');
```



i el paraboloid hiperbòlic

```
> plot3d([u+v,u-v,u*v],u=-4..4,v=-4..4,orientation=[60,75]);
```



4.2 Derivades parcials

Aquesta forma de definir una superfície, permet calcular de forma general les derivades parcials de la funció respecte cada paràmetre, així com les derivades segones

```
> fu:=(superficie,u,v)->diff(superficie(u,v),u) :
> fv:=(superficie,u,v)->diff(superficie(u,v),v) :
> fuu:=(superficie,u,v)->diff(superficie(u,v),u$2) :
> fuv:=(superficie,u,v)->diff(diff(superficie(u,v),u),v) :
> fvv:=(superficie,u,v)->diff(superficie(u,v),v$2) :
```

Exemple 1 Trobeu les derivades parcials de primer i segon ordre de la superfície $\Phi(u, v)$

```
> fu(Phi,u,v);fv(Phi,u,v);
      [eu cos(v), eu sin(v), 2u + 2]
      [-eu sin(v), eu cos(v), 0]

> fuu(Phi,u,v);fuv(Phi,u,v);fvv(Phi,u,v);
      [eu cos(v), eu sin(v), 2]
      [-eu sin(v), eu cos(v), 0]
      [-eu cos(v), -eu sin(v), 0]
```

4.3 Punts regulars i singulars d'una superfície

Recordem que un punt $(x_0, y_0, z_0) = \Phi(u_0, v_0)$ d'una superfície diferenciable és *regular* si es compleix que rang de $d\Phi_{(u_0, v_0)} = 2$ i que els punts no regulars de la superfície són els punts *singulars*.

Exemple 2 Trobeu els punts singulars de la superfície

$$\Phi_1(u, v) = \left(\frac{1 + \cos(2u)}{2} + \cos(u), \frac{\sin(2u)}{2} + \sin(u) + \frac{\sqrt{2}}{2}(v^2 - v), \frac{\sqrt{2}}{2}(v^2 + v) \right)$$

Primer de tot, la definim

```
> Phi1:=(u,v)->[(1+cos(2*u))/2+cos(u),
sin(2*u)/2+sin(u)+sqrt(2)*(v^2-v)/2,
sqrt(2)*(v^2+v)/2]:
```

```
> fu1:=fu(Phi1,u,v);fv1:=fv(Phi1,u,v);
```

$$fu1 := [-\sin(2u) - \sin(u), \cos(2u) + \cos(u), 0]$$

$$fv1 := \left[0, \frac{1}{2}\sqrt{2}(2v-1), \frac{1}{2}\sqrt{2}(2v+1) \right]$$

Calculem $d\Phi_1$

```
> A:=array(1..2,1..3,[fu1,fv1]);
```

$$A := \begin{bmatrix} -\sin(2u) - \sin(u) & \cos(2u) + \cos(u) & 0 \\ 0 & \frac{1}{2}\sqrt{2}(2v-1) & \frac{1}{2}\sqrt{2}(2v+1) \end{bmatrix}$$

Calculem els menors i estudiem per a quins valors s'anul·len

```
> m1:=A[1,1]*A[2,2]-A[1,2]*A[2,1];
```

```
m2:=A[1,1]*A[2,3]-A[1,3]*A[2,1];
```

```
m3:=A[1,2]*A[2,3]-A[2,1]*A[1,3];
```

$$m1 := \frac{1}{2}(-\sin(2u) - \sin(u))\sqrt{2}(2v-1)$$

$$m2 := \frac{1}{2}(-\sin(2u) - \sin(u))\sqrt{2}(2v+1)$$

$$m3 := \frac{1}{2}(\cos(2u) + \cos(u))\sqrt{2}(2v+1)$$

```
> solve(m1);solve(m2);solve(m3);
```

$$\{u = 0, v = v\}, \{u = \pi, v = v\}, \left\{u = \frac{2}{3}\pi, v = v\right\}, \left\{u = -\frac{2}{3}\pi, v = v\right\}, \left\{u = u, v = \frac{1}{2}\right\}$$

$$\{u = 0, v = v\}, \{u = \pi, v = v\}, \left\{u = \frac{2}{3}\pi, v = v\right\}, \left\{u = -\frac{2}{3}\pi, v = v\right\}, \left\{u = u, v = \frac{-1}{2}\right\}$$

$$\{u = \pi, v = v\}, \left\{v = v, u = \frac{1}{3}\pi\right\}, \left\{u = u, v = \frac{-1}{2}\right\}$$


El menors s'anul·len simultàniament per a $u = \pi$. Per tant els punts singulars de la superfície són

```
> sing:=simplify(subs(u=Pi,v=v,Phi1(u,v)));
```

$$sing := \left[0, \frac{\sqrt{2}}{2}(v^2 - v), \frac{\sqrt{2}}{2}(v^2 + v)\right]$$

que corresponen a una corba en el pla $x = 0$

```
> spacecurve([0, 1/2*sqrt(2)*v*(v-1), 1/2*sqrt(2)*v*(v+1)]
,v=-2*Pi..2*Pi,axes=frame);
```



4.4 La 1a i 2ona Forma Quadràtica Fonamental

Calculem ara la 1a i 2ona Forma Quadràtica Fonamental, donem en primer lloc un procediment general i després l'apliquem al cas de la superfície $\Phi(u, v)$.

Primera Forma Quadràtica Fonamental, **IForma**

```

> E:=(superficie,u,v)->
simplify(evalm(fu(superficie,u,v)*fu(superficie,u,v))):

> F:=(superficie,u,v)->
simplify(evalm(fu(superficie,u,v)*fv(superficie,u,v))):

> G:=(superficie,u,v)->
simplify(evalm(fv(superficie,u,v)*fv(superficie,u,v))):

> IForma:=(superficie,u,v)->
[[E(superficie,u,v),F(superficie,u,v)],
 [F(superficie,u,v),G(superficie,u,v)]]:

```

Per a calcular la Segona Forma Quadràtica Fonamental, cal trobar el vector normal unitari, que en diem N_s , a la superfície en un punt qualsevol. Aquest càlcul es fa utilitzant el procediment que es descriu a continuació i que, com se sap, consisteix en calcular el producte vectorial dels vectors d'una base de l'espai tangent i després dividir-lo per la norma.

```

> Ns:=proc(superficie,u,v) local w, vn, nn;
vn:=crossprod(fu(superficie,u,v),fv(superficie,u,v));
nn:=norm(vn,2); w:=[vn[1]/nn,vn[2]/nn,vn[3]/nn]; end:

```

La Segona Forma Quadràtica Fonamental, **IIForma**

```

> e:=(superficie,u,v)->
simplify(evalm(Ns(superficie,u,v)*fu(superficie,u,v))):

> f:=(superficie,u,v)->
simplify(evalm(Ns(superficie,u,v)*fv(superficie,u,v))):

> g:=(superficie,u,v)->
simplify(evalm(Ns(superficie,u,v)*fvv(superficie,u,v))):

> IIForma:=(superficie,u,v)->
[[e(superficie,u,v),f(superficie,u,v)],
 [f(superficie,u,v),g(superficie,u,v)]]:

```

Exemple 3 Calculeu la 1a i 2a Formes Quadràtiques Fonamentals de la superfície $\Phi(u, v)$ en el punt $\Phi(0, 0)$.

La 1a Forma Quadràtica Fonamental de $\Phi(u, v)$ en el punt $\Phi(0, 0)$ és

```

> I0:=simplify(subs(u=0,v=0,IForma(Phi,u,v)));
I0 := [[5, 0], [0, 1]]

```


El vector normal unitari a la superfície en el punt $\Phi(0, 0)$.

```
> Ns0:=simplify(subs(u=0,v=0,Ns(Phi,u,v)));
```

$$Ns0 := \left[-\frac{2}{5}\sqrt{5}, 0, \frac{1}{5}\sqrt{5}\right]$$

La 2a Forma Quadràtica Fonamental de $\Phi(u, v)$ en el punt $\Phi(0, 0)$.

```
> II0:=simplify(subs(u=0,v=0,IIForma(Phi,u,v))) ;
```

$$II0 := \left[[0, 0], \left[0, \frac{2}{5}\sqrt{5}\right]\right]$$

4.5 Angle entre corbes

L'angle que formen dues corbes α, β definides sobre una superfície S i que es tallen en un punt $p = \alpha(t_0) = \beta(t_1)$, és $\cos(\psi) = \frac{\alpha'(t_0)I_p\beta'(t_1)}{\|\alpha'(t_0)\|\|\beta'(t_1)\|}$ on I_p és la 1a Forma Quadràtica Fonamental de la superfície en el punt p . En particular l'angle ψ que formen les corbes coordenades és $\cos(\psi) = \frac{F}{\sqrt{EG}}$.

Exemple 4 Trobeu l'angle que formen les corbes $\alpha(t) = \Phi(t, t)$ i $\beta(t) = \Phi(t, 2t)$ en el punt corresponent a $t = 0$.

```
> a1:=simplify(subs(u=0,v=0,
evalm([1,1]&*IIForma(Phi,u,v)&*[1,2])/
sqrt(evalm([1,1]&*IIForma(Phi,u,v) &*[1,1]))/
sqrt(evalm([1,2]&*IIForma(Phi,u,v)&*[1,2]))));
```

$$a1 := \frac{7}{18}\sqrt{6}$$

Exemple 5 Calculeu l'angle que formen les corbes coordenades del paraboloid hiperbòlic en el punt $(1, 0, 1)$.

En aquest punt és $u = 1, v = 1$, per tant, si l'angle que formen és ψ , es compleix que $\cos(\psi) =$

```
> subs(u=1,v=1,F(parabhiper,u,v)/
sqrt(E(parabhiper,u,v)*G(parabhiper,u,v)));
```

$$\frac{1}{9}\sqrt{9}$$

Exemple 6 *Comproveu que la parametrització de la superfície $\Phi(u, v)$ és ortogonal.*

Només cal observar que $F = 0$ en qualsevol punt.

```
> IForma(Phi, u, v);
[[e(2u) + 4u2 + 8u + 4, 0], [0, e(2u)]]
```

4.6 Longitud d'una corba definida sobre una superfície


El següent procediment permet calcular la longitud aproximada d'una corba definida sobre una superfície. Els valors t_0 i t_1 corresponen als extrems de l'arc de corba de la qual es vol calcular la longitud. En molts casos la integral que resulta no té primitiva i per això es calcula numèricament fent servir `evalf`. El procediment consta de dos passos, primer de tot es calcula l'integrand i finalment s'avalua la integral. Proveu primer si és possible calcular la integral treient 'evalf' i si no, calculeu-la numèricament.

```
> Longitud:= proc(t0,t1,t,superficie,ucorba,vcorba)
local norma;
norma:=simplify(sqrt(subs(u=ucorba,v=vcorba,
E(superficie,u,v))*diff(ucorba,t)^2+ 2*subs(u=ucorba,v=vcorba,
F(superficie,u,v))*diff(ucorba,t)*diff(vcorba,t) +
subs(u=ucorba,v=vcorba,G(superficie,u,v))*diff(vcorba,t)^2));
evalf(int(norma,t=t0..t1)); end:
```

Exemple 7 *Calculeu la longitud de la corba $\Phi(t, t)$ definida sobre la superfície Φ , per a $-1 < t < 1$.*

Dibuixem la superfície i la corba

```
> corba:=spacecurve(Phi(t,t),t=-1..1,color=black,
thickness=2);
superf:=plot3d(Phi(u,v),u=-1.2..1.2,v=0..2*Pi,
orientation=[42,73]);
display(corba,superf);
```



```
> Longitud(-1,1,t,Phi,t,t);
5.229424745
```

4.7 Àrea d'una superfície

Aquest procediment per a calcular l'àrea d'una superfície és semblant al definit en el cas del càlcul de la longitud i s'ha de tenir en compte el mateix raonament pel que fa al càlcul de les integrals. Vegeu el capítol 2 corresponent a integració en vàries variables.

```
> Area:= proc(u0,u1,v0,v1,superficie,u,v) local inarea;
inarea:=simplify( sqrt(det(IForma(superficie,u,v)))));
evalf(int(int(inarea,u=u0..u1),v=v0..v1)); end:
```

Exemple 8 Calculeu l'àrea de la superfície esfèrica definida segons la parametrització donada a la secció 4.1.

Notem que els paràmetres que s'han pres per a definir-la són la latitud u i la longitud v i que el domini de definició és $u \in [0, 2\pi)$ i $v \in [-\pi/2, \pi/2]$. La integral que cal resoldre és:

```
> Int(Int(sqrt(simplify(det(IForma(esfera,u,v)) )),
v=-Pi/2..Pi/2),u=0..2*Pi);
```

$$\int_0^{2\pi} \int_{-1/2\pi}^{1/2\pi} \sqrt{R^4 \cos(v)^2} dv du$$

que, en aquest cas, es pot calcular de forma immediata

```
> assume(R>0):int(int(sqrt(simplify(det(
  IForma(esfera,u,v))))), v=-Pi/2..Pi/2),u=0..2*Pi);
      4 π R2
```

i fent servir el procediment per al cas general

```
> Area(0,2*Pi,-Pi/2,Pi/2,esfera,u,v);
      12.56637062 R2
```

4.8 Curvatura normal

La curvatura normal \mathbf{Kn} a una superfície en un punt p de la superfície segons la direcció $(u1, v1)$, s'obté

```
> Kn:=(u1,v1)->
  evalm([u1,v1]&*IIF&*[u1,v1])/evalm([u1,v1]&*IF&*[u1,v1]):
```

on IF i IIF són la 1a i 2a Forma Quadràtica Fonamental en el punt p .

Exemple 9 Calculeu la curvatura normal a la superfície Φ , definida a l'apartat 4.1, en el punt $(1, 2)$.

```
> IF:=subs(u=1,v=2,IForma(Phi,u,v)):
> IIF:=subs(u=1,v=2,IIForma(Phi,u,v)):
> simplify(Kn(u1,v1));
```

$$-2 \frac{e^2 (u1^2 - 2 v1^2)}{\sqrt{16 e^2 + e^4 (u1^2 e^2 + 16 u1^2 + v1^2 e^2)}}$$

4.9 Direccions principals i curvatures

Les curvatures principals d'una superfície en un punt p , es calculen com s'ha vist durant el curs, resolent l'equació que s'obté a l'igualar a zero la següent funció **CurvaturaPrincipal**,

```
> CurvaturaPrincipal:=(u0,v0,superficie,u,v)->
  evalf(subs(u=u0,v=v0,
```

```
(E(superficie,u,v)*G(superficie,u,v)-F(superficie,u,v)^2)*x^2-
(E(superficie,u,v)*g(superficie,u,v)+ G(superficie,u,v)*
e(superficie,u,v)-2*F(superficie,u,v)*f(superficie,u,v))* x+
(e(superficie,u,v)*g(superficie,u,v)-f(superficie,u,v)^2))):
```

Mentre que les direccions principals es troben resolent el següent sistema d'equacions (que té rang 1, com a molt), per als valors de k que corresponen a les curvatures trobades abans

```
> eq1:=(k,u0,v0,superficie)->subs(u=u0,v=v0,
(k* F(superficie,u,v)-f(superficie,u,v))*y+
(k*E(superficie,u,v)-e(superficie,u,v))*x):
eq2:=(k,u0,v0,superficie)->subs(u=u0,v=v0,
(k*F(superficie,u,v)-f(superficie,u,v))*x
+(k*G(superficie,u,v)-g(superficie,u,v))*y):
```

Exemple 10 Calculeu les curvatures i les direccions principals de la superfície $\Phi_1(u, v)$, definida en l'exemple 2 d'aquest mateix capítol, en el punt $(\pi/7, 2\pi/7)$.

Fem servir la funció *CurvaturaPrincipal* definida en el punt $(\pi/7, 2\pi/7)$ per a la superfície $\Phi_1(u, v)$

```
> CurvaturaPrincipal(Pi/7,2*Pi/7,Phi1,u,v);
15.31978210 x^2 + 9.798962444 x - 1.788968767
```

les seves arrels són les curvatures principals de $\Phi_1(u, v)$ en $(\pi/7, 2\pi/7)$

```
> sols:=solve(CurvaturaPrincipal(Pi/7,2*Pi/7,Phi1,u,v));
sols := -.7878483127, .1482202567
```

```
> Km:=sols[1];KM:=sols[2];
Km := -.7878483127
KM := .1482202567
```

resolent el sistema d'equacions que s'obté en fer $eq1=0$, $eq2=0$ obtenim les direccions principals. Notem que el sistema té rang menor que 2 i que en alguns casos potser convindrà escriure el sistema abans de resoldre'l per a evitar possibles divisions per zero (cas que el coeficient de x o de y s'anul·li). Aquest no és el cas de l'exercici que estem resolent i per això escrivim *solve* directament. La direcció principal per a la curvatura principal KM és

```
> solve(eq1(KM,Pi/7,2*Pi/7,Phi1),y);
solve(eq2(KM,Pi/7,2*Pi/7,Phi1),y);
-27.10189724 x
```

$$-27.10189682x$$

La direcció principal corresponent a la curvatura KM és $(1, -27.1)$.

I per a la curvatura Km

```
> solve(eq1(Km,Pi/7,2*Pi/7,Phi1),y);
solve(eq2(Km,Pi/7,2*Pi/7,Phi1),y);
      -.1710534920x
      -.1710534950x
```

la direcció principal és $(1, -0.17)$

4.10 Curvatura geodèsica

Donada una superfície en general, el problema de trobar-ne les corbes geodèsiques suposa resoldre un problema d'equacions diferencials de segon ordre. En aquesta secció comprovarem si una corba definida sobre una superfície és o no, geodèsica. Una corba és geodèsica si el vector normal a la superfície té la mateixa direcció que el vector normal a la corba.

Exemple 11 La corba $\beta(t) = \Phi(t, 0) = (e^t, 0, (t+1)^2)$ definida per $t > 0$, sobre la superfície $\Phi(u, v)$, és geodèsica?

Calculem el vector normal a la corba

```
> beta:=simplify(subs(u=t,v=0,Phi(u,v)));
beta1:=diff(beta,t);beta2:=diff(beta1,t);
      beta := [e^t, 0, (t + 1)^2]
      beta1 := [e^t, 0, 2t + 2]
      beta2 := [e^t, 0, 2]
```

el vector tangent és en la direcció

```
> tn:=beta1;
      tn := [e^t, 0, 2t + 2]
```

el vector binormal és en la direcció

```
> nib:=crossprod(beta1,beta2);
      nib := [0, (2t + 2) e^t - 2 e^t, 0]
```

i el vector normal en la direcció

```
> ron:=crossprod(nib,tn):
nnor:=[simplify(ron[1]),simplify(ron[2]),simplify(ron[3])];
      nnor := [4 e^t t (t + 1), 0, -2 e^(2t) t]
```

El vector normal a la superfície és

```
> NNor:=simplify(subs(u=t,v=0,Ns(Phi,u,v)));
```

$$NNor := \left[-2 \frac{(t+1)e^t}{\sqrt{e^{2\Re(t)} |2t+2|^2 + e^{4\Re(t)}}}, 0, \frac{e^{2t}}{\sqrt{e^{2\Re(t)} |2t+2|^2 + e^{4\Re(t)}}} \right]$$

Per tal de saber si la corba és geodèsica només cal observar que els vectors $NNor$ i $nnor$ tenen la mateixa direcció (hem de trobar un λ diferent de 0 tal que $nnor = \lambda \cdot NNor$):

```
> -2*t*sqrt(exp(2*Re(t))*(2*t + 2)^2 + exp(4*Re(t)))*NNor;
      [4 e^t t (t + 1), 0, -2 e^(2t) t]
```

per tant:

$$nnor = -2t \sqrt{e^{2\Re(t)} |2t+2|^2 + e^{4\Re(t)}} \cdot NNor$$

El coeficient $-2t \sqrt{e^{2\Re(t)} |2t+2|^2 + e^{4\Re(t)}}$ no s'anul·la per $t > 0$. Com que el vector normal de la corba β i el de la superfície Φ tenen la mateixa direcció, la corba és geodèsica

4.11 Projeccions i parametritzacions

En aquesta secció volem mostrar que les parametritzacions de les superfícies es poden veure com a inverses de les projeccions (en especial les cartogràfiques); per exemple la projecció de Mercator ve donada per

$$u = RS\lambda$$

$$v = RS \ln \tan(\pi/4 + \phi/2),$$

on R és el radi equatorial de la Terra, S és el factor d'escala del mapa i ϕ i λ , la latitud i la longitud respectivament. Si tenim una corba $\alpha(t) = (u(t), v(t))$ sobre el mapa, com la veuríem en realitat sobre l'esfera terrestre, això ens obliga a calcular la inversa de la projecció.

Sabem que a partir de la longitud i la latitud, l'esfera es parametritza com

$$(R \cos(\phi) \cos(\lambda), R \cos(\phi) \sin(\lambda), R \sin(\phi))$$

per tant si expressem ϕ, λ en funció dels altres paràmetres, obtenim la parametrització inversa de la projecció de Mercator:

```
> phi:=(v,R,S)->2*arctan(exp(v/R/S))-Pi/2;
lambda:=(u,R,S)->u/R/S;
```


$$\phi := (v, R, S) \rightarrow 2 \arctan(e^{\frac{v}{RS}}) - \frac{1}{2} \pi$$

$$\lambda := (u, R, S) \rightarrow \frac{u}{RS}$$

```
> InvMercator:=(u,v,R,S)->
[R*cos(phi(v,R,S))*cos(lambda(u,R,S)),
R*cos(phi(v,R,S))*sin(lambda(u,R,S)), R*sin(phi(v,R,S))]:
```

Com a aplicació suposeu que tenim una trajectòria rectilínia $\alpha(t) = (t, t)$ per a $t \in [0, 1]$ sobre el mapa d'escala $S = 1/2$, i voleu visualitzar-la sobre la superfície esfèrica de radi $2m$ i calcular-ne la longitud real.

```
> rec:=spacecurve(InvMercator(t,t,2,1/2),
t=0..1, thickness=3, color=black,scaling=constrained):
es:=plot3d(subs(R=2, esfera(u,v)),u=0..2*Pi,v=-Pi/2..Pi/2,
scaling=constrained):
display(es,rec);
```

La longitud per a $R = 2m$ és

```
> evalf(subs(R=2,Longitud(0,1,t,esfera,t,t)));
```

2.622884996

4.12 Exercicis proposats

1. Busqueu dues parametritzacions diferents del con de radi R i altura h (la primera, a partir de l'expressió del con com a superfície donada implícitament; i la segona, pensat com a superfície de revolució).
2. Busqueu una parametrització de l'el·lipsoide de semieixos a , a , c pensat com a superfície de revolució al voltant de l'eix x .
3. Busqueu una parametrització del tor de radi tubular a i radi intern r .
4. Considereu la superfície donada per la parametrització

$$\varphi(u, v) = \left(\frac{4u}{u^2 + v^2 + 4}, \frac{4v}{u^2 + v^2 + 4}, \frac{2(u^2 + v^2)}{u^2 + v^2 + 4} \right).$$

Demostreu que si prenem $(u, v) \in \mathbb{R}^2$, aquesta superfície és una esfera menys un punt. Quin?

5. Considereu la següent parametrització de la catenoide

$$\varphi(u, v) = (a \cosh(v) \cos(u), a \cosh(v) \sin(u), a v),$$

obtinguda en fer girar la catenària al voltant de l'eix OZ. Recordeu que les funcions cosinus hiperbòlic i sinus hiperbòlic es poden definir com

$$\cosh x = \frac{e^x + e^{-x}}{2}; \quad \sinh x = \frac{e^x - e^{-x}}{2},$$

i que $\cosh' x = \sinh x$ i $\cosh^2 x - \sinh^2 x = 1$.

6. Considereu la superfície donada per la parametrització

$$\varphi(u, v) = \left(u, \frac{v}{\sqrt{1+k^2 u^2}}, \frac{k u v}{\sqrt{1+k^2 u^2}} \right).$$

Demostreu si prenem $(u, v) \in \mathbb{R}^2$, aquesta superfície és el paraboloid hiperbòlic d'equacions $z = x y$.

7. Considereu la superfície donada per la parametrització

$$\varphi(u, v) = \left(\frac{u}{\sqrt{1+u^2+v^2}}, \frac{-v}{\sqrt{1+u^2+v^2}}, \frac{1}{\sqrt{1+u^2+v^2}} \right).$$

Demostreu que si prenem $(u, v) \in \mathbb{R}^2$, aquesta superfície és una semiesfera. Ara bé: conté o no conté l'equador, aquesta semiesfera?

8. Considereu la superfície donada per la parametrització

$$\varphi(u, v) = \left(\left(2 - v \sin \frac{u}{2}\right) \sin u, \left(2 - v \sin \frac{u}{2}\right) \cos u, v \cos \frac{u}{2} \right).$$

De quina superfície es tracta si $0 < u < 2\pi$ i $-1 < v < 1$?

9. Considereu la *superfície d'Enneper*, que ve donada per la parametrització

$$\varphi(u, v) = \left(u - \frac{u^3}{3} + u v^2 v - \frac{v^3}{3} + v u^2, u^2 - v^2 \right).$$

10. La *tractriu* és una corba plana que ve parametritzada per $(x(u), y(u)) = (\sin u, \cos u + \ln \tan u/2)$. La superfície de revolució al voltant de l'eix OZ que se n'obté s'anomena la *pseudoesfera*. Trobeu la seva parametrització.

11. Considereu la superfície donada per la parametrització

$$\varphi(u, v) = (\cos u - v \sin u, \sin u + v \cos u, v).$$

Demostreu que si prenem $(u, v) \in \mathbb{R}^2$, aquesta superfície és l'hiperboloide de revolució $x^2 + y^2 - z^2 = 1$.

12. Busqueu una parametrització de la superfície de revolució obtinguda al fer girar la gràfica de $y = \pi/2 + \arctan x$ al voltant de l'eix OX.

13. Considereu la superfície donada per la parametrització

$$\varphi(u, v) = \left(u, \frac{v}{k}, uv \right).$$

Demostreu si prenem $(u, v) \in \mathbb{R}^2$, aquesta superfície és el paraboloid hiperbòlic d'equacions $z = xy$.

14. Considereu la superfície donada per la parametrització

$$\varphi(u, v) = (u \cos(v/2), u \sin(v/2), e^{-v}).$$

15. Considereu la superfície donada per la parametrització

$$\varphi(u, v) = (\cos^3 u + v, \sin u \cos^2 u + 2v, v).$$

Digueu de quin tipus de superfície es tracta.

16. Busqueu una parametrització de la superfície de revolució obtinguda al fer girar la gràfica de $z = \cos^2 y$ al voltant de l'eix OY.

17. Busqueu una parametrització de l'el·lipsoide de semieixos a, b, b pensat com a superfície de revolució al voltant de l'eix x .

18. Busqueu la parametrització de la superfície de revolució obtinguda al fer girar al voltant de l'eix OZ, la corba parametritzada per $(x(u), z(u)) = (1/\cosh u, u - \tanh u)$ al pla xz (vegeu les definicions de les funcions hiperbòliques a l'apartat 5).

19. Considereu la *conoide de Plücker*, parametritzada per

$$\varphi(u, v) = (v \cos u, v \sin u, \cos u \sin u),$$

amb $u \in [0, 2\pi)$ i $v \in \mathbb{R}$.

20. Considereu la *superfície de Scherk*, parametritzada per

$$\varphi(u, v) = (\arccos u, \arccos v, \ln u - \ln v).$$

Determineu sobre quin domini $U \subset \mathbb{R}^2$ està definida.

21. Trobeu la inversa de la projecció gnomònica equatorial

$$u = R \tan(\lambda) \quad v = R \tan(\phi) / \cos(\lambda)$$

22. Localitzeu els punts singulars (no regulars) de la superfície, que a partir d'ara li direm S .
23. Doneu les expressions de la I i la II formes quadràtiques fonamentals de S en un punt $p = \varphi(u, v)$ qualsevol.
24. Preneu quatre corbes coordenades damunt el pla (u, v) de manera que formin un quadrat; per uniformitzar, que siguin $H_1 = (t, 0)$, $H_2 = (t, 1)$, $V_1 = (0, t)$, $V_2 = (1, t)$. Al quadrat en qestió li direm $Q \subset \mathbb{R}^2$, i denotarem per $\Omega = \varphi(Q) \subset S$ la seva imatge damunt la superfície. Calculeu:
- (a) L'angle que formen H_1 i V_1 . És ortogonal la parametrització?
 - (b) El perímetre d' Ω .
 - (c) L'àrea d' Ω .

Lliçó 5

Variable complexa

5.1 Operacions elementals amb nombres complexos.

En aquesta secció es presenta seqüencialment, a mode de pinzellades, la manera de dur a terme les operacions elementals que ens han semblat més significatives. Us aconsellem una lectura lineal de la secció deguda la seva futura utilitat.

Per començar, cal deixar clar que el nombre imaginari pur $i = \sqrt{-1}$ en MapleV s'indica amb la lletra i en majúscula:

```
> I*I;
```

-1

Fins a aquest capítol, ens hem acostumat al MapleV treballant només amb variables reals i no ens hem vist amb la necessitat de declarar el tipus de variable.

Veurem com ara les coses canvien. Fixeu-vos en aquest exemple:

```
> conjugate(a+b*I);
```

$\overline{a + Ib}$

Tots sabem que si a i b són reals, el seu conjugat és $a - ib$. Com és, doncs, que el MapleV (que de tants esculls ens treu!) no sigui capaç de dur a terme una operació tan fàcil? Vegem el següent exemple i en treurem l'entrellat:

Exemple 12 *Declaració de variables.*

> assume(a,real): assume(b,real): conjugate(a+b*I);

$$\bar{a} - I \bar{b}$$

És a dir, que si declarem explícitament que a i b són reals, llavors fa l'operació tal com ens esperàvem, malgrat que en lloc d' a i b ens escriu \bar{a} i \bar{b} per indicar-nos que les variables a i b són d'un determinat tipus.

D'altra banda, si a i b prenen valors concrets, no hi hauria cap problema:

> conjugate(2+3*I);

$$2 - 3I$$

Hem explorat, doncs, l'operació que més diferencia els complexos dels reals: la conjugació.

A part d'aquests detalls, no obstant, la resta d'operacions funciona com en els reals:

> expand((a1+a2*I)+(b1+b2*I));

$$a1 + I a2 + b1 + I b2$$

> expand((a1+a2*I)*(b1+b2*I));

$$a1 b1 + I a1 b2 + I a2 b1 - a2 b2$$

> expand((a1+a2*I)/(b1+b2*I));

$$\frac{a1}{a1 + I b2} + \frac{I a2}{a1 + I b2}$$

Observeu que la divisió no la racionalitza.

Hem vist també al llarg dels capítols anteriors que les comandes d'avaluació eren diverses (`eval`, `evalf`, `evalm`,...). Si volem que en fer les avaluacions es tingui en compte que les operacions són a \mathbb{C} , cal que emprem `evalc`. Per exemple:

Exemple 13 Ús d'`evalc`.

> evalc(log(3 + 5*I));

$$\frac{1}{2} \ln(34) + I \arctan\left(\frac{5}{3}\right)$$

> evalc(f(exp(a1+b1*I)));

$$f(e^{a1} \cos(b1) + I e^{a1} \sin(b1))$$

```
> evalc(2^(1+I));
                2 cos(ln(2)) + 2 I sin(ln(2))

> evalc(conjugate(exp(I)));
                cos(1) - I sin(1)
```

Les funcions que és capaç de simplificar la comanda `evalc` (algunes d'elles encara no s'han explicat) estan llistades a la Taula 5.1.

<code>sin</code>	<code>cos</code>	<code>tan</code>	<code>csc</code>	<code>sec</code>	<code>cot</code>
<code>sinh</code>	<code>cosh</code>	<code>tanh</code>	<code>csch</code>	<code>sech</code>	<code>coth</code>
<code>arcsin</code>	<code>arccos</code>	<code>arctan</code>	<code>arccsc</code>	<code>arcsec</code>	<code>arccot</code>
<code>arcsinh</code>	<code>arccosh</code>	<code>arctanh</code>	<code>arccsch</code>	<code>arcsech</code>	<code>arccoth</code>
<code>exp</code>	<code>ln</code>	<code>sqrt</code>	<code>^</code>	<code>abs</code>	<code>conjugate</code>
<code>polar</code>	<code>argument</code>	<code>signum</code>	<code>csgn</code>	<code>Re</code>	<code>Im</code>

Taula 5.1: Funcions que poden ser simplificades amb la comanda `evalc`.

Una altra operació elemental és l'obtenir la part real i la imaginària d'un nombre complex. Per exemple,

```
> Re(c+d*I);
                ℜ(c + I d)
```

no dóna c perquè no hem declarat c i d com a reals. En canvi, fixe'u-vos que si les declarem com a tals, l'operació es realitza de la manera esperada:

```
> assume(c,real);assume(d,real);Re(c+d*I);Im(c+d*I);
                c~
                d~
```

Una altra manipulació habitual en complexos és passar de forma cartesiana o binomial a forma polar. Hi ha, bàsicament, 3 funcions amb aquesta finalitat: `abs`, `argument`, `polar`.

```
> abs(a+b*I);
                √{a~2 + b~2}
```

```
> argument(a+b*I); argument(2+3*I);
      argument(a~ + I b~)
      arctan( $\frac{3}{2}$ )
```

Per emprar polar, però, cal carregar la llibreria corresponent:

Exemple 14 *Forma polar d'un nombre complex.*

Carreguem la llibreria:

```
> readlib(polar):
```

Calculem la forma polar d'un nombre complex qualsevol:

```
> polar(a+b*I);
      polar( $\sqrt{a^2 + b^2}$ , argument(a~ + I b~))
```

I la forma polar d'un nombre concret:

```
> polar(3*I);
      polar(3,  $\frac{1}{2}\pi$ )
```

Observeu en els resultats de sortida que si escrivim $\text{polar}(r, \theta)$ ens estem referint al nombre complex que té aquestes coordenades polars. El següent exemple ho il.lustra:

```
> polar(3, Pi/2)*polar(2, -Pi/2);
      polar(3,  $\frac{1}{2}\pi$ ) polar(2,  $-\frac{1}{2}\pi$ )
> simplify(%);
      polar(6, 0)
```

Per passar de polars a cartesianes, podeu utilitzar aquesta expressió que ens fabriquem a mida:

```
> cartesiana:=proc(r,theta) local x,y,z; x:=r*cos(theta);
y:=r*sin(theta); z:=x+I*y; end:
```

Ara la utilitzem per passar a cartesiana el nombre $z = 3_{\pi/2}$:

```
> cartesiana(3, Pi/2);
```


5.2 Funcions complexes. Holomorfia i conformitat.

Les funcions $f : \mathbb{C} \rightarrow \mathbb{C}$ no ofereixen dificultats especials des del punt de vista de sintaxi de MapleV respecte les funcions reals de variable real. Hi ha, és clar, diferències substancials des del punt de vista teòric: no és el mateix l'exponencial complexa que l'exponencial real, per exemple. Per defecte, en aquests casos, el MapleV aplica la definició a \mathbb{R} ; si es vol operar a \mathbb{C} cal fer ús de la comanda `evalc`, tal com hem explicat anteriorment. Observeu aquesta diferència en el següent exemple (hi continuem tenint a i b declarades com a reals; vegeu Secció 5.1):

Exemple 15 *Tractament diferenciat d'una funció segons el tipus de variable.*

En les dues properes comandes de MapleV es pot veure que si no manem fer un `evalc`, no s'aplica la definició habitual de la funció exponencial complexa.

```
> z:=a+b*I: eval(exp(z));
                                e^(a~+I b~)

> z:=a+b*I: evalc(exp(z));
                                e^a~ cos(b~) + I e^a~ sin(b~)
```

La llista de les funcions complexes que reben un tractament especial amb `evalc` l'hem inclosa a la Taula 5.1. D'altra banda, a través del Help també la podeu trobar a `evalc, functions`.

Abans de continuar, farem dissabte i esborrarem l'assignació de la variable z , que en aquests moments és $z = a + bi$, amb a i b reals:

```
> readlib(unassign): unassign('z');
```

Per acabar de visitar les principals tasques que haurem de realitzar en l'estudi de funcions complexes, desenvolupem una sèrie de qüestions (canvis de coordenades, holomorfia, conformitat,...) a tall d'exemple:

Exemple 16 *Canvis de coordenades.*

- Com s'escriu $f(z) = z^3 \bar{z}$ en coordenades cartesianes (x, y) , on $z = x + iy$?
- Com s'escriu $f(x, y) = x^4 - y^4 + i(2x^3y + 2xy^3)$ en funció de z i \bar{z} ?

a) Preparem una funció `Axy` que efectuï el procés de passar a coordenades x , y :

```
> Axy:=proc(f,z,w) subs(z=x+I*y,w=x-I*y,f(z,w)); end:
```

I ara l'apliquem a la funció proposada:

```
> f:=(z,w)->z^3*w:
```

```
> fxy:=expand(Axy(f,z,w));
```

$$fxy := x^4 + 2Ix^3y + 2Ixy^3 - y^4$$

És molt important fer un `expand` en lloc d'un `simplify`, ja que amb el `simplify` no distingiríem la part real de la imaginària.

Si, a més, volem separar-les, podem substituir I per zero a l'expressió que hem batejat com `fxy`, i obtindrem la part real:

```
> subs(I=0,expand(fxy));
```

$$x^4 - y^4$$

Multiplicant per $-i$ tota l'expressió aconseguim que la part imaginària passi a ser real i després podem substituir I per zero altra vegada. Obtenim, doncs, la part imaginària:

```
> subs(I=0,expand(-I*Axy(f,z,w)));
```

$$2x^3y + 2xy^3$$

NOTA: Ja sabem que $i = \sqrt{-1}$, però en aquesta substitució no avaluem el nombre imaginari i , si no que el considerem com un símbol.

b) Preparem una funció `Azzbarra` que ens passi a coordenades z , \bar{z} :

```
> Azzbarra:=proc(f,x,y)
```

```
subs(x=(z+conjugate(z))/2,y=(z-conjugate(z))/(2*I),f(x,y));
```

```
end:
```

Definim la funció:

```
> f:=(x,y)->x^4-y^4+I*(2*x^3*y+2*x*y^3):
```

I li apliquem `Azzbarra`:

```
> simplify(Azzbarra(f,x,y));
```

$$z^2 |z|^2$$

Observeu que ens hauria de donar la mateixa funció que a l'apartat a). Aparentment no és així, és a dir, no dona $z^3 \bar{z}$. Ara bé, com que $z\bar{z} = |z|^2$, llavors $z^3 \bar{z} = z^2 |z|^2$. Això posa de manifest la bona simplificació que ha fet el MapleV.

5.2.1 Funcions holomorfes

Exemple 17 *Estudi de les propietats d'holomorfa*

Són holomorfes les funcions $f_1(z) = \bar{z}$ i $f_2(z) = z\bar{z}$?

Per respondre-ho, cal veure si les funcions en qüestió compleixen les equacions de Cauchy-Riemann i, a més, si les seves derivades parcials són contínues. Aquest darrer aspecte no és calculístic i, per tant, l'ordinador ens hi pot ajudar ben poc. En canvi, comprovar les equacions de Cauchy-Riemann és ben factible. La condició que cal imposar és que la derivada respecte \bar{z} sigui nul·la ($f_{\bar{z}} = 0$).

De tota manera, per a MapleV, `conjugate(z)` no pot ser una variable. Us aconsellem, en aquests casos, introduir una variable artificial `w`, que per a nosaltres serà el conjugat i que oportunament ja substituïrem. Així:

```
> f1:=(z,w)->w:
> f2:=(z,w)->z*w:
```

La derivada de la primera funció respecte `w`...

```
> diff(f1(z,w),w);
```

1

és diferent de zero i, per tant, la primera funció no és holomorfa.

La segona, en canvi,...

```
> diff(f2(z,w),w);
```

z

i només s'anul·la a $z = 0$. Per tant, és l'únic punt on és holomorfa.

Exemple 18 *Més holomorfa.*

És holomorfa la funció $f(x, y) = x^2 - y^2 + 2xyi$?

Per resoldre aquest exercici tenim dues alternatives:

- utilitzar les equacions de Cauchy-Riemann en coordenades cartesianes;
- deixar $f(x, y)$ en funció de z i \bar{z} i després veure que $f_{\bar{z}} = 0$.

El farem de les dues maneres.

- Les equacions de Cauchy-Riemann en coordenades cartesianes són $u_x = v_y$, $v_x = -u_y$, entenent que $f(x, y) = u(x, y) + i v(x, y)$. En el nostre exemple:

```
> u:=(x,y)->x^2-y^2:  v:=(x,y)->2*x*y:
> simplify(diff(u(x,y),x)-diff(v(x,y),y));
0
> simplify(diff(u(x,y),y)+diff(v(x,y),x));
0
```

Efectivament, és holomorfa.

- Per transformar a les variables z i \bar{z} , utilitzem la funció Azzbarra que hem introduït a l'Exemple 15.

```
> f:=(x,y)->x^2-y^2+I*(2*x*y):
> simplify(Azzbarra(f,x,y));
z^2
```

Això ens indica clarament que f és holomorfa perquè no depèn de \bar{z} .

5.2.2 Transformacions conformes

Exemple 19 Transformacions conformes

És conforme la transformació $w = z^5 - 3z + 1$?

Suposant que és holomorfa (i aquesta funció ho és), només cal veure on s'anul·la la derivada respecte z :

```
> df:=diff(z^5-3*z+1,z);
df := 5 z^4 - 3
```

Li ordenem ara que ens solucioni $f'(z) = 0$ o, dit d'una altra manera, que calculi les arrels quartes de $\frac{3}{5}$:

```
> solve(df,z);
```

$$\frac{1}{5} \sqrt{5} 15^{1/4}, -\frac{1}{5} \sqrt{5} 15^{1/4}, \frac{1}{5} \sqrt{-5 \sqrt{15}}, -\frac{1}{5} \sqrt{-5 \sqrt{15}}$$

Si volem tenir una idea del seu valor, podem demanar-ne una aproximació numèrica:

```
> evalf(%);
```

$$.8801117368, -.8801117368, .8801117368 I, -.8801117368 I$$

Per tant, la transformació $w = z^5 - 3z + 1$ és conforme arreu excepte en aquests quatre punts.

Als darrers exemples hem estudiat la conformitat d'una transformació, concepte equivalent a la conservació d'angles. Queden encara dues qüestions per tractar respecte als angles format per dues corbes: 1) com es mesuren, 2) com es transformen sota l'acció d'una funció no conforme. La mesura es pot dur a terme per comparació dels arguments dels vectors tangents i no porta excessiva complicació quant a l'ús del manipulador algebraic. La segona qüestió, en canvi, creiem que mereix un tractament a part, que trobareu a la Secció 5.3. D'altra banda, també és interessant conèixer en què es transformen certes regions sota una transformació, aspecte que tractarem a la Secció 5.4.

5.3 Tractament gràfic de corbes paramètriques. Angles entre corbes.

Les corbes a \mathbb{C} vénen donades de la forma $z(t)$, i s'entén que es representen al pla, en les coordenades $\operatorname{Re}(z)$ i $\operatorname{Im}(z)$. Construïrem una funció estàndard per representar qualsevol corba en el pla complex.

```
> DibCorbaC := (z, t0, t1, c) -> plot([Re(z(t)), Im(z(t)), t=t0..t1], color=c);
```

$$\text{DibCorbaC} := (z, t0, t1, c) \rightarrow \text{plot}([\Re(z(t)), \Im(z(t)), t = t0..t1], \text{color} = c)$$

NOTA: No cal prendre's aquesta manera de tractar les corbes com l'única possible. Aquesta és una definició feta a mida per les necessitats d'aquest text, però hom pot muntar altres maneres de realitzar organitzadament els gràfics.

Vegem un exemple d'aplicació de la definició que hem donat:

Exemple 20 Representació gràfica d'una corba.


Representeu gràficament la corba $z(t) = \sin(t + it)$, amb $t \in [1, 3]$.

Primer definim la funció $z(t)$:

```
> z:=t->evalc(sin(t*I+t)):
```

i a continuació hi apliquem la comanda DibCorbaC que hem creat abans:

```
> DibCorbaC(z(t),-Pi,Pi,blue);
```



Tot i que en el format imprès no es pot distingir, mantenim l'opció de color. Quan executeu el programa en pantalla, es distingiran molt millor els angles que estiguen observant.

Sabem que si una funció f és conforme en z_0 , aleshores els angles entre corbes que passen per aquest punt es mantenen sota la transformació $w = f(z)$. D'altra banda, si f no és conforme en z_0 però admet un desenvolupament en sèrie de potències $f(z) - f(z_0) = a_m(z - z_0)^m + o(m)$, amb $a_m \neq 0$, llavors qualsevol angle que dues corbes formen en z_0 , es multiplica per un factor $m + 1$. En el proper exemple es recullen uns quants casos que il·lustren aquests fets; utilitzant la funció DibCorbaC els podem visualitzar.

Exemple 21 Transformacions d'angles.


Veurem com es transforma l'angle de 45° que les corbes $z_1(t) = t$ i $z_2(t) = t + ti$ formen en $z_0 = 0$ sota diverses funcions: una de conforme i dues de no conformes en aquest punt.

Primer de tot, definim les dues corbes i les dibuixem:

```
> z1:=t->evalc(t): z2:=t->evalc(t+t*I):

> with(plots):

> display(DibCorbaC(z1(t),-1,1,blue), DibCorbaC(z2(t),-1,1,red),
axes=none);
```



- Considerem la funció $f(z) = z(z + 1)$.


Comprovem primer que és conforme en $z = 0$:

```
> fa:=z->z*(z+1):
> subs(z=0,diff(fa(z),z));
```

1

Ara dibuixem les imatges:

```
> display(DibCorbaC(fa(z1(t)),-1,1,blue),
DibCorbaC(fa(z2(t)),-1,1,red),axes=none,scaling=constrained);
```



Observeu que l'angle continua essent de 45° . De cara a la visualització és important mantenir la mateixa escala per a les x i per a les y ; per això usem l'opció `scaling=constrained`.

- Considerem la funció $fb(z) = z^3 \sin z$.

Comprovem primer que NO és conforme en $z = 0$:


```
> fb:=z->z^3*sin(z):
> subs(z=0,diff(fb(z),z));
0
```

Busquem el desenvolupament en sèrie i observem que comença amb el terme de grau 3. Així, doncs, l'angle es quadruplicarà.

```
> series(diff(fb(z),z),z=0,5);
4 z^3 + O(z^5)
```

Ara dibuixem les imatges:

```
> display(DibCorbaC(fb(z1(t)),-1,1,blue),
DibCorbaC(fb(z2(t)),0,1,red),axes=none,scaling=constrained);
```

Efectivament, l'angle és de 180° .

- Considerem la funció $fc(z) = z - \sin z$.

Comprovem primer que NO és conforme en $z = 0$:

```
> fc:=z->z-sin(z):
> eval(subs(z=0,diff(fc(z),z)));
0
```

Busquem el desenvolupament en sèrie i observem que comença amb el terme de grau 2. Així, doncs, l'angle es triplicarà.


```
> series(diff(fc(z),z),z=0,5);

$$\frac{1}{2}z^2 - \frac{1}{24}z^4 + O(z^5)$$

```

Ara dibuixem les imatges:

```
> display(DibCorbaC(fc(z1(t)),-1,1,blue),
DibCorbaC(fc(z2(t)),-1,1,red),axes=none,scaling=constrained);
```



Efectivament, l'angle és de 135° .

L'altra part realment interessant de les transformacions és observar els canvis de forma que impliquen en regions determinades. La propera secció està dedicada a la visualització d'aquests canvis.

5.4 Imatges de dominis per transformacions conformes.

En aquesta secció ens ocuparem únicament de visualitzar les imatges de quadrilàters sota les transformacions elementals bàsiques. Les definicions, òbviament, es poden adaptar a d'altres tipus de regions. Utilitzarem la funció `DibCorbaC` que hem introduït més amunt per definir nous processos que dibuixin directament el quadrilàter en qüestió i la seva imatge. Marcarem cada costat amb un color diferent per poder seguir bé la regió imatge.


```
> DibRectC:=proc(x0,x1,x2,x3) local c0,c1,c2,c3,d0,d1,d2,d3;
c0:=x0+t*(x1-x0); c1:=x1+t*(x2-x1); c2:=x2+t*(x3-x2); c3:=x3+t*(x0-x3);
d0:=DibCorbaC(c0,0,1,blue); d1:=DibCorbaC(c1,0,1,red);
d2:=DibCorbaC(c2,0,1,black); d3:=DibCorbaC(c3,0,1,green);
display(d0,d1,d2,d3); end:
```

```
> DibImatge:=proc(f,x0,x1,x2,x3) local c0,c1,c2,c3,d0,d1,d2,d3;
c0:=f(x0+t*(x1-x0)); c1:=f(x1+t*(x2-x1));
c2:=f(x2+t*(x3-x2)); c3:=f(x3+t*(x0-x3));
d0:=DibCorbaC(c0,0,1,blue); d1:=DibCorbaC(c1,0,1,red);
d2:=DibCorbaC(c2,0,1,black); d3:=DibCorbaC(c3,0,1,green);
display(d0,d1,d2,d3); end:
```


Els paràmetres de `DibRectC` són les coordenades dels vèrtexs del quadrilàter. La comanda `DibImatge`, a més a més, duu la funció, f , com a paràmetre. A continuació, i a mode d'exemple, podeu veure com s'apliquen a les imatges de les funcions exponencial, sinus i logaritme, respectivament. En el cas del logaritme s'observa clarament el seu caràcter multiforme.

Exemple 22 *Imatge de quadrilàters per algunes transformacions elementals.*


```
> DibRectC(-.5-.5*I,-.5+.5*I,.5+.5*I,.5-.5*I);
```




```
> DibImatge(exp,-.5-.5*I,-.5+.5*I,.5+.5*I,.5-.5*I);
```



```
> DibImatge(sin,-.5-.5*I,-.5+.5*I,.5+.5*I,.5-.5*I);
```



> DibImatge(log, -.5-.5*I, -.5+.5*I, .5+.5*I, .5-.5*I);



5.5 Exercicis proposats.

1. Considereu la següent funció de variable complexa $f(z) = z + \bar{z}^2$.
 - (a) Expressen f en la forma $u(x, y) + i v(x, y)$; és a dir, busqueu les funcions u i v corresponents.
 - (b) És holomorfa, f ?
2. Expressen en forma cartesiana el nombre $\log i$; destaqueu-ne el valor principal.

3. Expressiu en funció de z i \bar{z} les funcions $u(x, y) + i v(x, y)$.

$$(a) \quad u(x, y) = \frac{x^2 + y^2 + x}{x^2 + y^2}, \quad v(x, y) = \frac{-y}{x^2 + y^2}.$$

$$(b) \quad u(x, y) = x + 1, \quad v(x, y) = y - 1.$$

4. Sigui $f(z) = u(x, y) + i v(x, y)$ una funció holomorfa tal que $u(x, y) = x^2 + 4x - y^2 + 2y$. Trobeu $v(x, y)$ i $f(z)$.

5. Trobeu gràficament la regió imatge del semiplà $y > 0$ sota la transformació $w = (1 + i)z$.

6. Demostreu que la imatge del semiplà $x < c_1$ (amb $c_1 < 0$) sota la transformació $w = 1/z$ és l'interior d'un cercle. Quina és la imatge quan $c_1 = 0$?

7. Trobeu la transformació racional lineal (TRL) que aplica els punts $z_1 = 2$, $z_2 = i$, $z_3 = -2$ sobre els punts $w_1 = 1$, $w_2 = i$, $w_3 = -1$.

8. Trobeu la TRL que aplica els punts $z_1 = \infty$, $z_2 = i$, $z_3 = 0$ sobre els punts $w_1 = 0$, $w_2 = i$, $w_3 = \infty$.

9. Dibuixeu la transformació de la regió rectangular $\Omega = \{-\pi \leq x \leq \pi, a \leq y \leq b\}$ per la funció $\sin z$.


10. Dibuixeu la imatge transformada del quadrat $R = [0, 1] \times [0, 1]$ sota l'aplicació $w = \cos z$?

11. Cerqueu per mètodes de variable complexa les trajectòries ortogonals de les següents famílies de corbes:

$$(a) \quad x^3 y - xy^3 = c.$$

$$(b) \quad e^{-x} \cos y + xy = c.$$

12. Intenteu aconseguir un dibuix similar al de la Figura 12 amb la funció $f(z) = 1/z$. Es tracta de prendre diferents quadrats estratègicament i dibuixar-los junt amb les seves imatges.



13. Definiu un procés específic per a dibuixar imatges de rectangles sota transformacions qualssevol del disc unitat en sí mateix. Haureu de deduir primer (o buscar a la bibliografia de l'assignatura) quina és l'expressió general d'aquestes transformacions.
14. Considereu la funció $f(z) = 1/z$. Demostreu que les circumferències i les rectes tenen per imatges circumferències o rectes. En particular, les circumferències que no passen per l'origen van a circumferències que no passen per l'origen; les circumferències que passen per l'origen van a rectes que no passen per l'origen, i viceversa; i les rectes que passen per l'origen, a rectes que passen per l'origen.

Indicacions: Les circumferències i les rectes es poden escriure de manera conjunta com corbes que compleixen $a(x^2 + y^2) + bx + cy + d = 0$, amb $b^2 + c^2 > 4ad$. Les rectes compleixen $a = 0$, i quan $d = 0$ la corba passa per l'origen.

Índex de comandes de MapleV

- abs, 67
- area, 55
- argument, 67
- assume, 29, 66

- conjugate, 66
- convert, 15, 26
- CurvaturaPrincipal, 56

- DEplot, 35
- DEtools, 34
- DibCorbaC, 73–75, 78
- DibImatge, 79
- DibRectC, 78
- diff, 26, 33
- display, 2, 3, 25, 27
- display3d, 10, 11
- dotprod, 26
- dsolve, 33

- Elgendre, 39
- esfera, 47
- evalc, 66, 75
- evalf, 73

- fieldplot, 20, 24, 25
- fieldplot3d, 21
- floor, 15

- grad, 21, 22
- grid, 20

- IForma, 51
- IIForma, 52

- implicitplot, 27, 38
- Int, 26, 29, 31
- int, 26, 29, 31
- inverse, 7

- jacobian, 20, 21, 23

- Kn, 56

- leastsqrs, 5, 15
- linalg, 1, 19, 41
- Longitud, 54

- map, 26
- matrix, 2, 5–7, 10, 13, 16

- Ns, 52

- pdesolve, 40
- plot, 25
- plot3d, 23, 27, 29
- plots, 1, 19, 41
- plottools, 1
- polar, 67
- polygonplot, 2, 3
- potential, 21, 22, 26
- poygonplot3d, 10, 11

- scaling, 76
- simplify, 29
- solve, 30, 73
- spacecurve, 27, 51
- stack, 14, 15
- subs, 14, 20, 22, 26, 27, 29, 30

textplot, 2, 3

textplot3d, 10, 11

unassign, 69

vector, 2-4, 10, 11, 13, 16, 17, 20-22, 24